HEWLETT-PACKARD

Step-by-Step Solutions
For Your HP Calculator

Engineering Applications



HP-32S

HEWLETT PACKARD

Help Us Help You!

Please take a moment to complete this postage-paid card, tear it out and put it in the mail. Your responses and comments will help us better understand your needs and will provide you with the best procedures to solve your problems. Thank you!

HELP US HELP YOU!

Book: **Engineering Applications** Date acquired: ____________________

Name ____________________

Street ____________________

City, State, Zip ____________________

Phone (______) ____________________ Business ______ or Home ______

1. What calculator will you use this book with?
   008 ☐ **HP-32S** 006 ☐ **Other** ____________________
2. How many other HP solution books have you bought for this calculator? ______
3. What is your OCCUPATION?
   101 ☐ **Student** 103 ☐ **Professional** 109 ☐ **Other** ____________________
4. Where did you purchase this book?
   403 ☐ **Bookstore** 404 ☐ **Discount or Catalog Store**
   407 ☐ **Mail Order** 410 ☐ **HP Direct** 411 ☐ **Other** ____________________
5. How did you first hear about this book?
   501 ☐ **HP Owner** 503 ☐ **Advertising** 506 ☐ **Salesperson** 507 ☐ **Brochure**
   508 ☐ **Other** ____________________
6. To what degree did this book influence your calculator purchase decision?
   601 ☐ **Major Influence** 602 ☐ **Minor Influence** 603 ☐ **No Influence**
7. How well does this book cover the material you expected?
   701 ☐ **Good** 702 ☐ **Moderate** 703 ☐ **Low**
8. What level of knowledge is required to make use of the topics in this book?
   801 ☐ **High** 802 ☐ **Medium** 803 ☐ **Low**
9. How clearly was the material in this book presented?
   901 ☐ **Good** 902 ☐ **Moderate** 903 ☐ **Low**
10. How would you rate the value of this book for your money?
    111 ☐ **High** 112 ☐ **Medium** 113 ☐ **Low**

Comments: (Please comment on improvements and additional applications or subjects you would like HP to cover in this or another solution book.) ____________________

NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

BUSINESS REPLY MAIL

FIRST CLASS MAIL PERMIT NO. 40 CORVALLIS, OR

POSTAGE WILL BE PAID BY ADDRESSEE

HEWLETT-PACKARD COMPANY
CALCULATOR MARKET RESEARCH
1000 NE CIRCLE BLVD.
CORVALLIS OR 97330-9988

Engineering Applications

Step-by-Step Solutions for Your HP-32S Calculator

HEWLETT PACKARD

Edition 1 June 1988
Reorder Number 00032-90057

Notice

Corvallis Division
1000 N.E. Circle Blvd.
Corvallis, OR 97330, U.S.A.

Printing History

Edition 1 June 1988 Mfg. No. 00032-90058

Contents

How to Use This Book

The *Engineering Applications* solutions book provides sets of keystrokes and routines to help you solve a variety of engineering, statistics, and mathematics problems. The routines have been written to provide for easy use and minimum memory space. This book is to be used with the HP-32S calculator.

Before you use the solutions in this book, you should be familiar with the following concepts from the owner's manual:

- The basics of your calculator — how to perform arithmetic operations, move from menu to menu, and use the menu keys to do calculations.
- How to use the SOLVE function to solve for a variable.
- How to enter numbers for statistics.
- How to key in and run a program. You may wish to refer to the Function Index in your *HP-32S Owner's Manual* for information on how to key a particular function into a program.
- How to determine the number of bytes in a program and how to display the checksum.

Keys and Menu Selection. A key on the calculator keyboard is represented like this: [STO] . A shifted function is preceded by a shift key, like this: ■ [ASIN] . A menu label is represented like this: {DSE} . It is often necessary to go through several menus to obtain the desired function. For example: ■ [TESTS] {x?y} {>y} .

Display Formats. The examples in this book show numbers displayed to four decimal places. You may change the number of decimal places your calculator displays by pressing ■ [DISP] {FX} and the number of decimal places desired. If you wish to see the full 12-digit precision of a number regardless of the display format, press ■ [SHOW] ; the full precision number is displayed as long as you hold down the [SHOW] key.

Programs. The HP-32S calculator uses single letters to denote program labels; you have up to 26 labels in program memory. When keying in the program listings in this book, your calculator will display a `DUPLICAT. LBL` error if you use a letter for a label that is already used in program memory. To avoid this problem, simply choose a different letter to designate the label. Be sure to change any [XEQ] or [GTO] statements that correspond to the newly-assigned label and make note of the changes so that when you execute the routine, you specify the proper label.

When you key in a number having more than three digits in its mantissa, the HP-32S automatically inserts appropriate commas into the number in both data-entry and programming modes.

Note

Changing the label name of a program affects its checksum.

Checksum. A checksum is provided for each program listing as a verification that the program has been keyed in correctly. To view the checksum, press ■ [MEM] {PGM} and scroll through the listing to the program label you want to check. Press and hold ■ [SHOW] to display the checksum.

Our thanks to Tony Vogt of Oregon State University for developing the problems and equations in this book.

1

Electrical Engineering

Reactance Chart

This program calculates the resonant frequency, the inductance, or the capacitance of an LC circuit at resonance given the other two variables. It also calculates the capacitive and inductive reactances at resonance, which are equal.

$$f = \frac{1}{2\pi\sqrt{LC}}$$

$$X = \frac{1}{2\pi f C}$$

where:

L = inductance in henrys.

C = capacitance in microfarads.

f = resonant frequency in hertz.

X = reactance in ohms.

Program Listing.

When keying in steps R12 and R13, press 6 [ENTER] [←] and [+/−]. These steps require less memory than keying in −6.

```
R01 LBL R           R12 6
R02 INPUT F         R13 +/-
R03 INPUT L         R14 10ˣ
R04 INPUT C         R15 ×
R05 2               R16 RCL× C
R06 π               R17 RCL× W
R07 ×               R18 1
R08 RCL× F          R19 -
R09 STO W           R20 RTN
R10 RCL× L          Checksum = 6CD2
R11 STO X
```

Flags Used. None.

Memory Required. 30 bytes.

Remarks. The value of *C* is in microfarads to increase the precision of the SOLVE function.

Program Instructions.

1. Key in program listing; press [C] when finished.
2. Press ■ [SOLVE/∫] {FN} R.
3. Specify the unknown variable by pressing ■ [SOLVE/∫] {SOLVE} *variable*.
4. Key in the variable value at each prompt and press [R/S].
5. See the variable for which the program is solving.
6. Press ■ [VIEW] X to see the reactance.
7. For a new case, go to step 3.

Variables Used.

L = inductance in henrys.

C = capacitance in microfarads.

F = resonant frequency in hertz.

X = reactance in ohms.

$W = 2\pi f$ (angular velocity ω in radians per second).

Example. Resonant Frequency and Reactance.
Calculate F and X, when $L = 1.0$ mh and $C = 0.25$ μf.

Keys:	Display:	Description:
■ [SOLVE/∫] {FN}	FN= _	Prompts for program label.
R	*value*	Specifies program R.
■ [SOLVE/∫] {SOLVE}	SOLVE _	Prompts for the unknown variable.
F	L?*value*	Starts program R; prompts for variables *except F*.
[E] 3 [+/−] [R/S]	C?*value*	*C* must be in microfarads.
.25 [R/S]	F=10,065.8424	Displays the resonant frequency.
■ [VIEW] X	X=63.2456	Displays the reactance.

Impedance of a Ladder Network

This program computes the input impedance of a ladder network. Elements are added one at a time from right to left. The first element must be parallel. The input impedance may be viewed at any point in the ladder as the elements are added.

Given an input impedance of Y_{in}, adding a shunt (parallel) R, L, or C results in a new input impedance of:

$$Y_{new} = \begin{cases} Y_{in} + \left(\frac{1}{R_p} + j0\right) \\ Y_{in} + \left(0 - j\frac{1}{\omega L_p}\right) \\ Y_{in} + (0 + j\omega C_p) \end{cases}$$

Adding a series R, L, or C, we have:

$$Y_{new} = \begin{cases} \left(\frac{1}{Y_{in}} + (R_s + j0)\right)^{-1} \\ \left(\frac{1}{Y_{in}} + (0 + j\omega L_s)\right)^{-1} \\ \left(\frac{1}{Y_{in}} + \left(0 - j\frac{1}{\omega C_s}\right)\right)^{-1} \end{cases}$$

where $Z = \frac{1}{Y}$ and $\omega = 2\pi f$.

Program Listing.

```
N01 LBL N
N02 INPUT F
N03 2
N04 x
N05 π
N06 x
N07 STO W
N08 0
N09 ENTER
N10 RTN
Checksum = 026C
S01 LBL S
S02 CMPLX1/x
S03 R↓
S04 R↓
S05 CMPLX1/x
S06 CMPLX+
S07 CMPLX1/x
S08 RTN
Checksum = 9EA3
P01 LBL P
P02 CMPLX+
P03 RTN
Checksum = 9583
C01 LBL C
C02 INPUT C
C03 RCLx W
C04 0
C05 RTN
Checksum = 10CA
L01 LBL L
L02 INPUT L
L03 RCLx W
L04 1/x
L05 +/-
L06 0
L07 RTN
Checksum = 517E
R01 LBL R
R02 0
R03 INPUT R
R04 1/x
R05 RTN
Checksum = F867
Z01 LBL Z
Z02 CMPLX1/x
Z03 y,x→θ,r
Z04 STO Z
Z05 x<>y
Z06 STO A
Z07 VIEW A
Z08 VIEW Z
Z09 x<>y
Z10 θ,r→y,x
Z11 CMPLX1/x
Z12 RTN
Checksum = 5450
```

Flags Used. None.

Memory Required. 75 bytes.

Remarks.

- The program performs calculations using the admittance in *cartesian* coordinates but displays the result as an impedance in *polar* coordinates.
- Angles must be consistent with the angular mode currently set in the calculator.

Program Instructions.

1. Key in the program listing and press [C] when finished.
2. Press [XEQ] N.
3. Key in the frequency and press [R/S] .
4. Select the appropriate element to add:
   - Press [XEQ] R to add a resistor.
   - Press [XEQ] L to add an inductor.
   - Press [XEQ] C to add a capacitor.
5. Key in the value at the prompt and press [R/S] .
6. Select the appropriate means of adding the element:
   - Press [XEQ] P to add the element in parallel.
   - Press [XEQ] S to add the element in series.
7. To add another element, go to step 4.
8. Press [XEQ] Z to see the angle of the input impedance.
9. Press [R/S] to see the magnitude of the input impedance.
10. Optional: press [R/S] to continue adding elements to the ladder.

Variables Used.

R = resistance in ohms.

L = inductance in henrys.

C = capacitance in farads.

Z = magnitude of the input impedance in ohms.

A = input impedance angle.

F = frequency in hertz.

W = $2\pi f$ (angular velocity ω in radians per second).

Example: RLC Ladder Network. Find the input impedance of the following circuit at a frequency of 1 MHz:



Keys:	**Display:**	**Description:**
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] N	F?*value*	Inputs frequency.
[E] 6 [R/S]	0.0000	
[XEQ] R	R?*value*	Adds resistor in parallel (first element must be in parallel).
100 [R/S]	0.0100	
[XEQ] P	0.0100	
[XEQ] C	C?*value*	Adds capacitor in series.
650 [E] 12 [+/−] [R/S]	0.0000	
[XEQ] S	0.0014	
[XEQ] L	L?*value*	Adds inductor in parallel.
120 [E] 6 [+/−] [R/S]	0.0000	
[XEQ] P	0.0014	
[XEQ] R	R? 100.0000	Adds resistor in parallel.
1000 [R/S]	0.0010	
[XEQ] P	0.0024	
[XEQ] Z	A= −41.8224	Displays the input impedance angle.
[R/S]	Z=306.7333	Displays the input impedance.

Smith Chart Conversions

The distance between a point on a Smith Chart and its center may be measured using a number of parameters. This program performs conversions between several of the most commonly used parameters: standing wave ratio, reflection coefficient, and return loss. It may also be used to convert between impedance and reflection coefficient.

σ = voltage standing wave ratio = $\dfrac{1+\rho}{1-\rho}$.

SWR = standing wave ratio expressed in decibels.

ρ = reflection coefficient.

R.L. = return loss.

These parameters are related as follows:

$$SWR = 20 \log \sigma$$

$$R.L. = 20 \log \frac{1}{\rho}$$

$$\sigma = \frac{1+\rho}{1-\rho}$$



These relationships are perhaps more clearly seen in this sketch:



For a system having characteristic impedance Z_0, the impedance and reflection coefficient are related by

$$\Gamma = \rho \measuredangle \phi = \frac{\frac{\mathbf{Z}}{Z_0} - 1}{\frac{\mathbf{Z}}{Z_0} + 1}$$

and

$$\mathbf{Z} = Z \measuredangle \theta = Z_0 \frac{1+\Gamma}{1-\Gamma}$$

where:

Γ = complex reflection coefficient.

$\rho = |\Gamma|$.

$\phi = \measuredangle \Gamma$.

$\mathbf{Z}$ = impedance.

$Z = |\mathbf{Z}|$.

$\theta = \measuredangle \mathbf{Z}$.

Program Listing.

```
A01 LBL A
A02 +/-
Checksum = 13CE
D01 LBL D
D02 20
D03 ÷
D04 10^x
D05 RTN
Checksum = B0E3
B01 LBL B
B02 STO K
B03 1
B04 +
B05 1
B06 RCL- K
B07 ÷
B08 RTN
Checksum = 68C2
E01 LBL E
E02 STO K
E03 1
E04 -
E05 1
E06 RCL+ K
E07 ÷
E08 RTN
Checksum = 17DB
F01 LBL F
F02 1/x
Checksum = 7789
C01 LBL C
C02 LOG
C03 20
C04 ×
C05 RTN
Checksum = 5C10
H01 LBL H
H02 1
H03 STO K
H04 INPUT M
H05 INPUT P
H06 INPUT D
H07 INPUT C
H08 RCL M
H09 RCL P
H10 θ,r→y,x
H11 RCL+ K
H12 RCL M
H13 RCL P
H14 +/-
H15 θ,r→y,x
H16 RCL+ K
H17 CMPLX÷
H18 RCL D
H19 RCL C
H20 θ,r→y,x
H21 CMPLX×
H22 y,x→θ,r
H23 STO Z
H24 x<>y
H25 STO T
H26 VIEW T
H27 VIEW Z
H28 RTN
Checksum = E61D
G01 LBL G
G02 INPUT T
G03 INPUT Z
G04 INPUT D
G05 INPUT C
G06 RCL T
G07 RCL Z
G08 θ,r→y,x
G09 RCL D
```

```
G10 RCL C
G11 θ,r→y,x
G12 CMPLX÷
G13 STO A
G14 x<>y
G15 STO B
G16 x<>y
G17 1
G18 -
G19 RCL B
G20 1
G21 RCL+ A
G22 CMPLX÷
G23 y,x→θ,r
G24 STO P
G25 x<>y
G26 STO M
G27 VIEW M
G28 VIEW P
G29 RTN
Checksum = 6425
```

Flags Used. None.

Memory Required. 130.5 bytes.

Remarks.

- Each routine is independent of the others. Therefore, key in only those routines that will be used.
- Angles must be consistent with the angular mode currently set in the calculator.

Program Instructions.

1. Key in the program listings of the routines to be used; press [C] when finished.
2. For conversions involving real number parameters (routines *A* thru *F*):
   - Key in the variable and select the appropriate routine:
     - Press [XEQ] A to convert *R.L.* to ρ.
     - Press [XEQ] B to convert ρ to σ.
     - Press [XEQ] C to convert σ to *SWR*.
     - Press [XEQ] D to convert *SWR* to σ.
     - Press [XEQ] E to convert σ to ρ.
     - Press [XEQ] F to convert ρ to *R.L.*
   - Optional: continue converting by executing the next routine in the sequence.

3. To convert from **Z** to Γ:
   - Press [XEQ] G.
   - Key in the values at each prompt and press [R/S] .
   - See *M*; press [R/S] ; see *P*.
4. To convert from Γ to **Z**:
   - Press [XEQ] H.
   - Key in the variables at each prompt and press [R/S] .
   - See *T*; press [R/S] ; see *Z*.
5. For a new case, go to step 2, 3, or 4.

Variables Used.

Z = magnitude of the impedance.

T = angle of the impedance.

P = magnitude of the complex reflection coefficient.

M = angle of the complex reflection coefficient.

C = magnitude of the characteristic impedance.

D = angle of the characteristic impedance.

A, K, B = variables used for intermediate results.

Example 1. Convert a 10 dB *SWR* to σ.

Keys:	Display:	Description:
10 [XEQ] D	3.1623	Displays σ.

Example 2. Convert a 5 dB return loss to *SWR*.

Keys:	Display:	Description:
5 [XEQ] A	0.5623	Displays ρ.
[XEQ] B	3.5698	Displays σ.
[XEQ] C	11.0528	Displays *SWR*.

Example 3. A 75 Ω system is terminated with an impedance of 53 at an angle of 41°. Find the reflection coefficient.

Keys:	Display:	Description:
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] G	T?*value*	Inputs values.
41 [R/S]	Z?*value*	
53 [R/S]	D?*value*	
0 [R/S]	C?*value*	
75 [R/S]	M=118.3651	Displays Γ.
[R/S]	P=0.4107	Displays ρ.

Example 4. A reflection coefficient of 0.35 at an angle of 11° is observed in a 100 Ω system. Find the impedance.

Keys:	Display:	Description:
[XEQ] H	M?*value*	Inputs values.
11 [R/S]	P?*value*	
.35 [R/S]	D?*value*	
0 [R/S]	C?*value*	
100 [R/S]	T=8.6547	Displays angle.
[R/S]	Z=203.8784	Displays magnitude of the impedance, Z.

Transistor Amplifier Performance

This program calculates several small-signal properties of a transistor amplifier given the h-parameter matrix and the source and load impedances. The properties computed are the current and voltage gains and the input and output impedances.



Definition of h-parameter matrix:

$$\begin{bmatrix} v_1 \\ i_2 \end{bmatrix} = \begin{bmatrix} h_i & h_r \\ h_f & h_o \end{bmatrix} \begin{bmatrix} i_1 \\ v_2 \end{bmatrix}$$

Current gain:

$$A_i = \frac{i_2}{i_1} = \frac{-h_f}{1 + h_o Z_L}$$

Voltage gain:

$$A_v = \frac{v_2}{v_1} = \frac{A_i Z_L}{Z_{in}}$$

Voltage gain with source resistor:

$$A_{vs} = \frac{v_2}{v_S} = \frac{A_i Z_L}{Z_{in} + Z_S}$$

Input impedance:

$$Z_{in} = h_i + h_r Z_L A_i$$

Output impedance:

$$Z_{out} = \frac{h_i + Z_S}{h_o h_i + h_o Z_S - h_f h_r}$$

Program Listing.

```
T01 LBL T
T02 CLVARS
T03 1.012
T04 STO i
Checksum = 7FE4
X01 LBL X
X02 INPUT(i)
X03 ISG i
X04 GTO X
X05 RCL I
X06 RCL J
X07 θ,r→y,x
X08 STO J
X09 x<>y
X10 STO I
X11 RCL G
X12 RCL+ K
X13 RCL H
X14 RCL× L
X15 θ,r→y,x
X16 1
X17 +
X18 CMPLX1/x
X19 RCL E
X20 RCL F
X21 +/-
X22 θ,r→y,x
X23 CMPLX×
X24 STO N
X25 x<>y
X26 STO M
X27 x<>y
X28 XEQ V
X29 RCL C
X30 RCL+ K
X31 RCL D
X32 RCL× L
X33 θ,r→y,x
X34 CMPLX×
X35 RCL A
X36 RCL B
X37 θ,r→y,x
X38 CMPLX+
X39 STO P
X40 x<>y
X41 STO O
X42 x<>y
X43 CMPLX1/x
X44 RCL M
X45 RCL N
```

```
X46 CMPLX×
X47 RCL K
X48 RCL L
X49 θ,r→y,x
X50 CMPLX×
X51 XEQ V
X52 RCL O
X53 RCL P
X54 CMPLX×
X55 RCL O
X56 RCL+ I
X57 RCL P
X58 RCL+ J
X59 CMPLX÷
X60 XEQ V
X61 RCL O
X62 RCL P
X63 XEQ V
X64 RCL G
X65 RCL H
X66 θ,r→y,x
X67 RCL I
X68 RCL J
X69 CMPLX×
X70 RCL G
X71 RCL+ A
X72 RCL H
X73 RCL× B
X74 θ,r→y,x
X75 CMPLX+
X76 RCL E
X77 RCL+ C
X78 RCL F
X79 RCL× D
X80 θ,r→y,x
X81 CMPLX-
X82 CMPLX1/x
X83 RCL A
X84 RCL B
X85 θ,r→y,x
X86 RCL+ J
X87 x<>y
X88 RCL+ I
X89 x<>y
X90 CMPLX×
Checksum = 72DD
V01 LBL V
V02 y,x→θ,r
V03 STO R
V04 x<>y
V05 STO T
V06 x<>y
V07 θ,r→y,x
V08 VIEW T
V09 VIEW R
V10 RTN
Checksum = EFAE
```

Flags Used. None.

Memory Required. 164 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- Angles must be consistent with the angular mode currently set in the calculator.
- To limit the number of variables, the program uses variable *T* for the angle and *R* for the magnitude of all of the output results.

Program Instructions.

1. Key in the program listings; press [C] when finished.
2. Press [XEQ] T.
3. Key in the variables at each prompt and press [R/S] .
4. See the angle of A_i and press [R/S] .
5. See the magnitude of A_i and press [R/S] .
6. See the angle of A_v and press [R/S] .
7. See the magnitude of A_v and press [R/S] .
8. See the angle of A_{vs} and press [R/S] .
9. See the magnitude of A_{vs} and press [R/S] .
10. See the angle of Z_{in} and press [R/S] .
11. See the magnitude of Z_{in} and press [R/S] .
12. See the angle of Z_{out} and press [R/S] .
13. See the magnitude of Z_{out}.
14. For a new case, go to step 2.

Variables Used.

A = angle of h_i.

B = magnitude of h_i.

C = angle of h_r.

D = magnitude of h_r.

E = angle of h_f.

F = magnitude of h_f.

G = angle of h_o.

H = magnitude of h_o.

I = angle of Z_{in}.

J = magnitude of Z_{in}.

K = angle of Z_{out}.

L = magnitude of Z_{out}.

T = angle of A_i, A_v, A_{vs}, Z_{in}, Z_{out}.

R = magnitude of A_i, A_v, A_{vs}, Z_{in}, Z_{out}.

N, M, O, P, i = variables used for intermediate results.

Example. Find the small-signal properties of a transistor that has the following h-parameter matrix with source and load impedances of 1000 and 5000 ohms, respectively.

$$h = \begin{bmatrix} 1000 & 150E-6 \\ 75 & 50E-6 \end{bmatrix}$$

Keys:	Display:	Description:
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] T	A?0.0000	
[R/S]	B? 0.0000	Inputs values.
1000 [R/S]	C? 0.0000	
[R/S]	D? 0.0000	
150 [E] 6 [+/-] [R/S]	E? 0.0000	
[R/S]	F? 0.0000	
75 [R/S]	G? 0.0000	
[R/S]	H? 0.0000	
50 [E] 6 [+/-] [R/S]	I? 0.0000	
[R/S]	J? 0.0000	
1000 [R/S]	K? 0.0000	
[R/S]	L? 0.0000	
5000 [R/S]	T=180.0000	Displays angle of A_i.
[R/S]	R=60.0000	Displays magnitude of A_i.
[R/S]	T=180.0000	Displays angle of A_v.
[R/S]	R=314.1361	Displays magnitude of A_v.
[R/S]	T=180.0000	Displays angle of A_{vs}.
[R/S]	R=153.4527	Displays magnitude of A_{vs}.
[R/S]	T=0.0000	Displays angle of Z_{in}.
[R/S]	R=955.0000	Displays magnitude of Z_{in}.
[R/S]	T=0.0000	Displays angle of Z_{out}.
[R/S]	R=22,535.2113	Displays magnitude of Z_{out}.

2

Mechanical Engineering

Black Body Thermal Radiation

All bodies emit thermal radiation according to their temperature. The higher the temperature, the more thermal radiation emitted. A black body is one that emits the maximum possible amount of energy at every wavelength for a specified temperature. The figure below represents the black body thermal emission as a function of wavelength.

This program can be used to calculate:

- The wavelength of maximum emissive power for a given temperature.
- The temperature corresponding to a particular wavelength of maximum emissive power.
- The total emissive power for all wavelengths.
- The emissive power at a particular wavelength and temperature.



$$\lambda_{max}T = c_3$$

$$E_{b(0-\chi)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

where:

λ_{max} = wavelength of maximum emissivity in microns.

T = absolute temperature in °R or K.

$E_{b(0-\chi)}$ = total emissive power in Btu/hr – ft^2 or watts/cm^2.

$E_{b\lambda}$ = emissive power at λ in Btu/hr – ft^2 – μm or watts/cm^2 – μm.

$c_1 = 1.8887982 \times 10^7$ Btu – μm^4/hr – ft^2 = 5.9544×10^3 Wμm^4/cm^2.

$c_2 = 2.58984 \times 10^4$ μm – °R = 1.4388×10^4 μm – K.

$c_3 = 5.216 \times 10^3$ μm – °R = 2.8978×10^3 μm – K.

$\sigma = 1.713 \times 10^{-9}$ Btu/hr – ft^2 – °R^4 = 5.6693×10^{-12} W/cm^2 – K^4.

$\sigma_{exp} = 1.731 \times 10^{-9}$ Btu/hr – ft^2 – °R^4 = 5.729×10^{-12} W/cm^2 – K^4.

Program Listing.

```
B01 LBL B
B02 INPUT W
B03 INPUT T
B04 RCL W
B05 1/x
B06 RCL× B
B07 RCL÷ T
B08 e^x
B09 1
B10 -
B11 1/x
B12 RCL W
B13 5
B14 y^x
B15 ÷
B16 2
B17 ×
B18 π
B19 ×
B20 RCL× A
B21 STO P
B22 VIEW P
B23 RTN
Checksum = 52E5
E01 LBL E
E02 INPUT T
E03 4
E04 y^x
E05 RCL× S
E06 STO E
```

```
E07 VIEW E
E08 RTN
Checksum = 15C6
W01 LBL W
W02 INPUT T
W03 RCL C
W04 RCL÷ T
W05 STO W
W06 VIEW W
W07 RTN
Checksum = F2C4
T01 LBL T
T02 INPUT W
T03 RCL C
T04 RCL÷ W
T05 STO T
T06 VIEW T
T07 RTN
Checksum = CA08
```

Flags Used. None.

Memory Required. 67.5 bytes.

Remarks. The values of the constants differ between sources.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Store the constants *A*, *B*, *C*, and *S* in the appropriate storage registers.
3. Select the appropriate routine:
   - Press [XEQ] W to calculate the wavelength of maximum power for a given temperature.
   - Press [XEQ] T to calculate the temperature corresponding to a particular wavelength of maximum power.
   - Press [XEQ] E to calculate the total emissive power.
   - Press [XEQ] B to calculate the emissive power at a particular wavelength.
   - Key in the variables at each prompt and press [R/S] .
   - See the variable for which the program is solving.
4. For a new case, go to step 3.

Variables Used.

T = temperature.

W = wavelength.

E = total emissive power.

P = emissive power at a given wavelength.

A = constant c_1 .

B = constant c_2 .

C = constant c_3 .

S = constant σ.

Example. If sunlight has a maximum wavelength of .550 μm, what is the sun's temperature in K? Assume the sun is a black body. What is the total emissive power and the emissive power at λ_{max}? What is the emissive power at λ = 0.400 μm (ultraviolet limit) and 0.700 μm (infrared limit)?

Keys:	Display:	Description:
5.9544 [E] 3 [STO] A 1.4388 [E] 4 [STO] B 2.8978 [E] 3 [STO] C 5.6693 [E] 12 [+/-] [STO] S		Stores constants.
[XEQ] T	W?*value*	
.55 [R/S]	T=5,268.7273	Displays temperature.
[XEQ] E	T?5,268.7273	
[R/S]	E=4,368.7009	Displays the total emissive power.
[XEQ] B	W?0.5500	Correct value already stored in *W*.

[R/S]	T?5,268.7273	
[R/S]	P=5,222.8745	Displays the emissive power.
[XEQ] B	W?0.5500	
.4 [R/S]	T?5,268.7273	
[R/S]	P=3,964.8581	Displays the emissive power.
[XEQ] B	W?0.4000	
.7 [R/S]	T?5,268.7273	
[R/S]	P=4,593.4033	Displays the emissive power.

Ideal Gas Equation

Many gases obey the ideal gas law at high temperatures and low pressures. This program calculates any one of the four variables of the ideal gas equation when the other three are known.

$$PV = nRT$$

where:

P = pressure.

V = volume.

n = number of moles.

R = Universal Gas Constant.

T = absolute temperature.

Table 2-1. Values of the Universal Gas Constant

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/M^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.08205	liter-atm/g mole-K	atm	liter/g mole	K
0.7302	atm-ft^3/lbm mole-°R	atm	ft^3/lbm mole	°R
10.73	psi-ft^3/lbm mole-°R	psi	ft^3/lbm mole	°R
1545	psf-ft^3/lbm mole-°R	psf	ft^3/lbm mole	°R

Program Listing.

```
G01 LBL G              G08 RCL× V
G02 INPUT P            G09 RCL N
G03 INPUT V            G10 RCL× R
G04 INPUT N            G11 RCL× T
G05 INPUT R            G12 -
G06 INPUT T            G13 RTN
G07 RCL P              Checksum = 63C5
```

Flags Used. None.

Memory Required. 19.5 bytes.

Remarks. Value of *R* must be compatible with units of *P*, *V*, and *T*.

Program Instructions.

1. Key in program listing; press [C] when finished.
2. Press ■ [SOLVE/∫] {FN} G, then specify the unknown variable by pressing ■ [SOLVE/∫] {SOLVE} *variable*.
3. Key in the variables at each prompt and press [R/S] .
4. See the variable for which the program is solving.
5. For a new case, go to step 2.

Variables Used.

P = absolute pressure.

V = volume.

N = number of moles present.

R = Universal Gas Constant.

T = absolute temperature.

Example 1: Pressure. If 1.2 moles of air are enclosed in 40,000 cm³ at 1500 K, what is the pressure in atmospheres?

Keys:	Display:	Description:
■ [SOLVE/∫] {FN}	FN= _	Prompts for program label.
G		Specifies program G.
■ [SOLVE/∫] {SOLVE}	SOLVE _	Prompts for the unknown variable.
P 40000 [R/S] 1.2 [R/S] 82.05 [R/S]	V?*value* N?*value* R?*value* T?*value*	Starts program G; prompts for variables *except P*.
1500 [R/S]	P=3.6923	Displays the pressure.

Example 2: Specific Volume. What is the specific volume (ft³/lbm) of a gas at a pressure of 3 atmospheres and a temperature of 540°R? The molecular weight is 32 lbm/lbm-mole.

Keys:	Display:	Description:
■ [SOLVE/∫] {SOLVE}	SOLVE _	Prompts for the unknown variable. (It is not necessary to redefine the program label being executed since it was defined in the last example.
V 3 [R/S] 32 [1/x] [R/S] .7302 [R/S]	P?*value* N?*value* R?*value* T?*value*	Starts program G; prompts for the variables *except V*.
540 [R/S]	V=4.1074	Displays specific volume.

Conduit Flow

This program solves for either the average velocity or the pressure drop for viscous, incompressible flow in conduits.

For laminar flow ($Re < 2300$):

$$f = 16/Re$$

For turbulent flow ($Re > 2300$):

$$V^2 = \frac{\Delta P/\rho}{2\left(f\frac{L}{D} + \frac{K_T}{4}\right)}$$

$$f = \frac{0.0772}{\left\{\log\left[\frac{6.9}{\text{Re}} + \left(\frac{\varepsilon}{3.7D}\right)^{1.111}\right]\right\}^2}$$

where:

V = average velocity.

ΔP = pressure drop.

L = conduit length.

D = conduit diameter. If the conduit is *not* circular, use an *equivalent* diameter defined by:

$$D_{eq} = 4 \times \frac{\textit{Cross Sectional Area}}{\textit{Wetted Perimeter}}$$

ε = surface irregularity.

Re = Reynolds number; $Re = DV/\nu$.

ν = fluid kinematic viscosity.

ρ = fluid density

f = Fanning friction factor.

K_T = sum of fitting factors.

Table 2-2. Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5 - 10
Angle valve, wide open	3.8
Gate valve, wide open	0.15 - 0.19
Gate valve, $^3/_4$ open	0.85
Gate valve, $^1/_2$ open	4.4
Gate valve, $^1/_4$ open	20
90° elbow	0.4 - 0.9
Standard 45° elbow	0.35 - 0.42
Tee, through side outlet	1.5
Tee, straight through	0.4
180° bend	1.6
Entrance to circular pipe	0.25 - 0.50
Sudden expansion	$(1 - A_{up}/A_{dn})^2$ *
Acceleration from $V = 0$ to $V = V_{entrance}$	1.0
*A_{up} is the upstream area and A_{dn} is the downstream area.	

Table 2-3. Surface Irregularities

Material	ε (Feet)	ε (Meters)
Drawn or smooth tubing	5.0×10^{6}	1.5×10^{6}
Commercial steel or wrought iron	1.5×10^{4}	4.6×10^{5}
Asphalted cast iron	4.0×10^{4}	1.2×10^{4}
Galvanized iron	5.0×10^{4}	1.5×10^{4}
Cast iron	8.3×10^{4}	2.5×10^{4}
Wood stave	6.0×10^{4} to 3.0×10^{3}	1.8×10^{4} to 9.1×10^{4}
Concrete	1.0×10^{3} to 1.0×10^{2}	3.0×10^{4} to 3.0×10^{3}
Riveted steel	3.0×10^{3} to 3.0×10^{2}	9.1×10^{4} to 9.1×10^{3}

Program Listing.

```
C01 LBL C
C02 INPUT E
C03 INPUT D
C04 INPUT V
C05 INPUT B
C06 INPUT L
C07 INPUT K
C08 INPUT S
C09 INPUT P
C10 RCL E
C11 RCL÷ D
C12 3.7
C13 ÷
C14 1.111
C15 y^x
C16 RCL V
C17 RCL× D
C18 RCL÷ B
C19 STO R
C20 2,300
C21 x>y?
C22 GTO L
C23 R↓
C24 1/x
C25 6.9
C26 ×
C27 +
C28 LOG
C29 3.6
C30 ×
C31 1/x
C32 x²
Checksum = 01E3
D01 LBL D
D02 STO F
D03 RCL× L
D04 RCL÷ D
D05 RCL K
D06 4
D07 ÷
D08 +
D09 2
D10 ×
D11 RCL V
D12 x²
D13 ×
D14 RCL P
D15 RCL÷ S
D16 -
D17 RTN
Checksum = 188C
L01 LBL L
L02 16
L03 RCL÷ R
L04 GTO D
L05 RTN
Checksum = F63D
```

Flags Used. None.

Memory Required. 121 bytes.

Program Instructions.

1. Key in program listing; press [C] when finished.
2. Press ■ [SOLVE/∫] {FN} C, then specify the unknown variable by pressing ■ [SOLVE/∫] {SOLVE} *variable*.
3. Key in the variables at each prompt and press [R/S] .
4. See the variable for which the program is solving.
5. Optional: Press ■ [VIEW] R to see the Reynolds number.
6. Optional: Press ■ [VIEW] F to see the Fanning friction factor.
7. For a new case, go to step 2.

Variables Used.

V = average velocity.

P = pressure drop.

L = conduit length.

D = conduit diameter.

E = surface irregularity.

R = Reynolds number.

B = fluid kinematic viscosity.

S = fluid density.

F = Fanning friction factor.

K = fitting coefficient.

Example: Pressure Drop. A 60-meter pipe has three 180 degree bends ($K_T = 3 \times 1.6$). The fluid is water($\nu = 9.3 \times 10^{-7}$ m^2/s, $\rho = 1000$ kg/m^3). The pipe diameter is 0.030 m and the surface roughness is 3×10^{-4} m. If the average velocity is 3.20 m/s, what is the pressure drop in Pascals? What is the Reynolds number? What is the Fanning friction factor?

Keys:	Display:	Description:
■ [SOLVE/∫] {FN}	FN= _	Prompts for program label.
C	*value*	Specifies program C.
■ [SOLVE/∫] {SOLVE}	SOLVE _	Prompts for the unknown variable.
P 3 [E] 4 [+/-] [R/S] .03 [R/S] 3.2 [R/S] 9.3 [E] 7 [+/-] [R/S] 60 [R/S] 4.8 [R/S]	E?*value* D?*value* V?*value* B?*value* L?*value* K?*value* S?*value*	Starts program C; prompts for variables *except P*.
[E] 3 [R/S]	P=418,351.2590	Displays the pressure drop.
■ [VIEW] R	R=103,225.8065	Displays the Reynolds number.
■ [VIEW] F	F=0.0096	Displays the friction factor.

Static Equivalent at a Point

This program calculates the two reaction forces necessary to balance any given two-dimensional force vectors, provided the vectors act through the same point. The direction of the reaction forces must be specified as an angle relative to an arbitrary axis.



Equations:

$$R_1 \cos\theta_1 + R_2 \cos\theta_2 = \sum F \cos\phi$$

$$R_1 \sin\theta_1 + R_2 \sin\theta_2 = \sum F \sin\phi$$

where:

F = magnitude of each known force.

ϕ = direction of each known force.

R_1 = first reaction force.

θ_1 = direction of R_1.

R_2 = second reaction force.

θ_2 = direction of R_2.

Program Listing.

```
S01 LBL S
S02 CLVARS
S03 INPUT N
Checksum = CC9D
A01 LBL A
A02 INPUT T
A03 INPUT F
A04 RCL T
A05 RCL F
A06 θ,r→y,x
A07 STO+ X
A08 x<>y
A09 STO+ Y
A10 DSE N
A11 GTO A
A12 INPUT A
A13 SIN
A14 STO A
A15 LASTx
A16 COS
A17 STO C
A18 INPUT B
A19 SIN
A20 STO B
A21 LASTx
A22 COS
A23 STO D
A24 RCL X
A25 RCL× B
A26 RCL D
A27 RCL× Y
A28 -
A29 RCL A
A30 RCL× D
A31 RCL C
A32 RCL× B
A33 -
A34 ÷
A35 STO R
A36 VIEW R
A37 LASTx
A38 RCL C
```

```
A39 RCLx Y              A44 ÷
A40 RCL X               A45 STO R
A41 RCLx A              A46 VIEW R
A42 -                   A47 RTN
A43 x<>y                Checksum = 665C
```

Flags Used. None.

Memory Required. 75 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- A positive value of force (tension) points away from the origin; a negative value (compression) points toward the origin.
- Angles must be consistent with the angular mode currently set in the calculator.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] S.
3. Key in the variables at each prompt and press [R/S] .
4. See the first reaction force, then press [R/S] .
5. See the second reaction force.
6. For a new case, go to step 2.

Variables Used.

N = number of known forces.

T = angle of each known force.

F = value of each known force.

A = direction of the first reaction force.

B = direction of the second reaction force.

R = value of the unknown forces R_1 and R_2 .

D, X, Y, C = variables used for intermediate results.

Example 1: Balancing a Single Vector. Find the reaction forces in the following diagram:



Keys:	**Display:**	**Description:**
[XEQ] S	N?0.0000	
1 [R/S]	T?0.0000	Inputs known values.
135 [R/S]	F?0.0000	
75 [R/S]	A?0.0000	
30 [R/S]	B?0.0000	
270 [R/S]	R=61.2372	Displays the first reaction force.
[R/S]	R=83.6516	Displays the second reaction force.

Example 2: Forces in a Bridge Truss. Find the reaction forces in structural members AE and CE. Assume pin connections at the joint.



Keys:	**Display:**	**Description:**
[XEQ] S	N?0.0000	
2 [R/S]	T?0.0000	Inputs known values.
45 [R/S]	F?0.0000	
100 [R/S]	T?45.0000	
180 [R/S]	F?100.0000	
120 [R/S]	A?0.0000	
[R/S]	B?0.0000	
135 [R/S]	R= −21.4214	Displays the first reaction force.
[R/S]	R= −100.0000	Displays the second reaction force.

Composite Section Properties

The mechanical properties of a constant cross section member composed of a finite number of rectangular elements can be computed by adding the contribution of each rectangular region individually. This program uses this principle to calculate the area of a section, the moments of inertia about the specified set of axes, the moments of inertia about an axis translated to the centroid, the moments of inertia of the principal axes, and the angle of rotation between the translated axes and the principal axes.

$$A_{s\,i} = \Delta x_i \Delta y_i$$

$$A = A_{s1} + A_{s2} + A_{s3} + \cdots + A_{sn}$$

$$\overline{x} = \frac{\sum_{i=1}^{n} x_{0i} A_{s\,i}}{A}$$

$$\overline{y} = \frac{\sum_{i=1}^{n} y_{0i} A_{s\,i}}{A}$$

$$I_{xy} = \sum_{i=1}^{n} x_{0i} y_{0i} A_{s\,i} \qquad I_{\overline{x}\,\overline{y}} = I_{xy} - A\,\overline{x}\,\overline{y}$$

$$I_x = \sum_{i=1}^{n} \left(y_{0i}^2 + \frac{\Delta y_i^2}{12} \right) A_{s\,i} \qquad I_{\overline{x}} = I_x - A\,\overline{y}^2$$

$$I_y = \sum_{i=1}^{n} \left(x_{0i}^2 + \frac{\Delta x_i^2}{12} \right) A_{s\,i} \qquad I_{\overline{y}} = I_y - A\,\overline{x}^2$$

$$J = I_x + I_y$$

$$\phi = \frac{1}{2} \tan^{-1} \left(\frac{2 I_{\overline{x}\,\overline{y}}}{I_{\overline{x}} - I_{\overline{y}}} \right)$$

$$I_{\overline{x}\phi} = I_{\overline{x}} \cos^2\phi + I_{\overline{y}} \sin^2\phi + I_{\overline{x}\,\overline{y}} \sin 2\phi$$

$$I_{\overline{y}\phi} = I_{\overline{y}} \cos^2\phi + I_{\overline{x}} \sin^2\phi + I_{\overline{x}\,\overline{y}} \sin 2\phi$$

$$J_\phi = I_{\overline{x}\phi} + I_{\overline{y}\phi}$$

where:

Δx_i = width of a rectangular element.

Δy_i = height of a rectangular element.

A_{si} = area of an element.

A = total area of the section.

$\overline{x}$ = x-coordinate of the centroid.

$\overline{y}$ = y-coordinate of the centroid.

x_{0i} = x-coordinate of the centroid.

y_{0i} = y-coordinate of the centroid.

I_x = moment of inertia about the x-axis.

I_y = moment of inertia about the y-axis.

J = polar moment of inertia about the origin.

I_{xy} = product of inertia about the origin.

$I_{\overline{x}}$ = moment of inertia about the x-axis translated to the centroid.

$I_{\overline{y}}$ = moment of inertia about the y-axis translated to the centroid.

$I_{\overline{xy}}$ = product of inertia about the translated axis.

ϕ = angle between the translated axis and the principal axis.

$I_{\overline{x}\phi}$ = moment of inertia about the principal x-axis.

$I_{\overline{y}\phi}$ = moment of inertia about the principal y-axis.

J_ϕ = polar moment of inertia about the principal axis.

Program Listing.

```
S01 LBL S
S02 CLVARS
S03 INPUT N
Checksum = CC9D
U01 LBL U
U02 INPUT X
U03 INPUT Y
U04 INPUT S
U05 INPUT T
U06 RCL× S
U07 STO B
U08 STO+ A
U09 RCL× Y
U10 STO+ D
U11 RCL X
U12 RCL× B
U13 STO+ C
U14 RCL Y
U15 x²
U16 RCL T
U17 x²
U18 12
U19 ÷
U20 +
U21 RCL× B
U22 STO+ H
U23 RCL X
U24 x²
U25 RCL S
U26 x²
U27 12
U28 ÷
U29 +
U30 RCL× B
U31 STO+ I
U32 +
U33 STO+ J
U34 RCL X
U35 RCL× Y
U36 RCL× B
U37 STO+ P
U38 DSE N
U39 GTO U
U40 RCL P
U41 RCL C
U42 RCL× D
U43 RCL÷ A
U44 -
U45 STO Q
U46 RCL C
U47 RCL÷ A
U48 VIEW A
U49 STO X
U50 VIEW X
U51 x²
U52 RCL D
U53 RCL÷ A
U54 STO Y
U55 VIEW Y
U56 x²
U57 RCL× A
U58 +/-
U59 RCL H
U60 VIEW H
U61 +
U62 STO H
U63 R↓
U64 RCL× A
U65 +/-
U66 RCL I
U67 VIEW I
U68 VIEW J
U69 VIEW P
U70 +
```

```
U71 STO I
U72 RCL H
U73 STO J
U74 VIEW H
U75 x<>y
U76 VIEW I
U77 STO+ J
U78 -
U79 STO D
U80 RCL Q
U81 STO P
U82 VIEW P
U83 x=0?
U84 GTO V
U85 ÷
U86 1/x
U87 2
U88 ×
U89 ATAN
U90 2
U91 ÷
U92 +/-
Checksum = A4B6
```

```
V01 LBL V
V02 STO G
V03 VIEW G
V04 2
V05 ×
V06 SIN
V07 RCL× P
V08 +/-
V09 RCL G
V10 SIN
V11 x²
V12 RCL× D
V13 -
V14 RCL+ H
V15 STO H
V16 VIEW H
V17 RCL- J
V18 +/-
V19 STO I
V20 VIEW I
V21 VIEW J
V22 RTN
Checksum = 63B7
```

Flags Used. None.

Memory Required. 175.5 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- For a given origin, the polar moment of inertia is constant regardless of the angular rotation. Therefore, $J_{\overline{xy}}$ is equal to J_{ϕ}.
- It is possible to obtain a negative value for the product of inertia.

Program Instructions.

1. Key in the program listing; press [C] when done.
2. Press [XEQ] S.
3. Key in the variables at each prompt and press [R/S].
4. See the results as they are displayed and press [R/S].
5. For a new case, go to step 2.

Variables Used.

$X = x_{0i}$ and $\overline{x}$.

$Y = y_{0i}$ and $\overline{y}$.

$S = \Delta x_i$.

$T = \Delta y_i$.

$H = I_x, I_{\overline{x}}$, and $I_{\overline{x}\phi}$.

$I = I_y, I_{\overline{y}}$, and $I_{\overline{y}\phi}$.

$J = J$ and J_ϕ.

$P = I_{xy}, I_{\overline{xy}}$.

A = total area of the entire section.

G = angle between the translated axis and the principal axis.

N = number of sections.

D, C, B, Q = variables used for intermediate results.

Example 1: Rectangular Section. Calculate the section properties of the following cross section:



Table of Inputs

x	y	Δx	Δy
2	1.5	4	3

Keys:	Display:	Description:
[XEQ] S	N?0.0000	
1 [R/S]	X?0.0000	Inputs known values.
2 [R/S]	Y?0.0000	
1.5 [R/S]	S?0.0000	
4 [R/S]	T?0.0000	
3 [R/S]	A=12.0000	Displays area.
[R/S]	X=2.0000	Displays $\overline{x}$.
[R/S]	Y=1.5000	Displays $\overline{y}$.
[R/S]	H=36.0000	Displays I_x .
[R/S]	I=64.0000	Displays I_y .
[R/S]	J=100.0000	Displays J .
[R/S]	P=36.0000	Displays I_{xy} .
[R/S]	H=9.0000	Displays $I_{\overline{x}}$.
[R/S]	I=16.0000	Displays $I_{\overline{y}}$.
[R/S]	P=0.0000	Displays $I_{\overline{xy}}$.
[R/S]	G=0.0000	Displays ϕ .
[R/S]	H=9.0000	Displays $I_{\overline{x}\phi}$.
[R/S]	I=16.0000	Displays $I_{\overline{x}\phi}$.
[R/S]	J=25.0000	Displays J_{ϕ} .

Example 2: Composite Section. Calculate the section properties of the following section:



Table of Inputs

	x	y	Δx	Δy
1	5	11.5	6	1
2	1	6	2	12
3	7	1	10	2

Keys:	Display:	Description:
[XEQ] S	N?0.0000	
3 [R/S]	X?0.0000	Inputs known values.
5 [R/S]	Y?0.0000	
11.5 [R/S]	S?0.0000	
6 [R/S]	T?0.0000	
1 [R/S]	X?5.0000	
1 [R/S]	Y?11.5000	
6 [R/S]	S?6.0000	
2 [R/S]	T?1.0000	
12 [R/S]	X?1.0000	
7 [R/S]	Y?6.0000	
1 [R/S]	S?2.0000	
10 [R/S]	T?12.0000	
2 [R/S]	A=50.0000	Displays area.
[R/S]	X=3.8800	Displays $\overline{x}$.
[R/S]	Y=4.6600	Displays $\overline{y}$.
[R/S]	H=1,972.6667	Displays I_x .
[R/S]	I=1,346.6667	Displays I_y .
[R/S]	J=3,319.3333	Displays J .
[R/S]	P=629.0000	Displays I_{xy} .
[R/S]	H=886.8867	Displays $I_{\overline{x}}$.
[R/S]	I=593.9467	Displays $I_{\overline{y}}$.
[R/S]	P= −275.0400	Displays $I_{\overline{xy}}$.
[R/S]	G=30.9814	Displays ϕ .
[R/S]	H=1,052.0261	Displays $I_{\overline{x}\phi}$.
[R/S]	I=428.8072	Displays $I_{\overline{x}\phi}$.
[R/S]	J=1,480.8333	Displays J_{ϕ} .

Soderberg's Equation for Fatigue

This program calculates any one of the six variables in Soderberg's equation for fatigue when the other five are known. Soderberg's equation is shown graphically in the figure below.



Equation:

$$\frac{s_{yp}}{FS} = \frac{s_{max} + s_{min}}{2} + K\left(\frac{s_{yp}}{s_e}\right)\left(\frac{s_{max} - s_{min}}{2}\right)$$

where:

s_{yp} = yield point stress.

s_e = endurance stress from reversed bending tests.

s_{max} = maximum applied stress.

s_{min} = minimum applied stress.

K = stress concentration factor.

FS = factor of safety.

Program Listing.

```
S01 LBL S
S02 INPUT Y
S03 INPUT E
S04 INPUT A
S05 INPUT B
S06 INPUT K
S07 INPUT F
S08 RCL Y
S09 RCL÷ F
S10 RCL A
S11 RCL+ B
S12 2
S13 ÷
S14 -
S15 RCL A
S16 RCL- B
S17 2
S18 ÷
S19 RCL× Y
S20 RCL÷ E
S21 RCL× K
S22 -
S23 RTN
Checksum = C95D
```

Flags Used. None.

Memory Required. 34.5 bytes.

Remarks.

- Soderberg's equation is valid for ductile materials only.
- Fatigue effects are magnified in corrosive environments.

Program Instructions.

1. Key in the program listing, pressing [C] when finished.
2. Press ■ [SOLVE/∫] {FN} S, then specify the unknown variable by pressing ■ [SOLVE/∫] {SOLVE} *variable*.
3. Key in the variables at each prompt and press [R/S] .
4. See the variable for which the program is solving.
5. For a new case, go to step 2.

Variables Used.

Y = yield point stress.

E = endurance stress.

A = maximum applied stress.

B = minimum applied stress.

K = stress concentration factor.

F = factor of safety.

Example. What is the maximum allowable applied stress if the minimum applied stress is 15,000 psi?

s_{yp} = 80,000 psi.

s_e = 30,000 psi.

K = 1.5.

FS = 2.0.

Keys:	Display:	Description:
■ [SOLVE/∫] {FN}	FN= _	Prompts for program label.
S	*value*	Specifies program S.
■ [SOLVE/∫] {SOLVE}	SOLVE _	Prompts for the unknown variable.
A 80000 [R/S] 30000 [R/S] 15000 [R/S] 1.5 [R/S]	Y?*value* E?*value* B?*value* K?*value* F?*value*	Starts program S; prompts for variables *except A*.
2 [R/S]	A=25,000.0000	Displays the maximum applied stress.

3

Civil Engineering

Mohr's Circle for Stress

This program calculates the 2-D Mohr's circle for stress from equiangular or rectangular strain gage data or directly from known stresses.

Configuration Code	**1**	**2**
Type of Rosette	Rectangular	Delta (Equiangular)
		
Principal Strains ϵ_1, ϵ_2	$\frac{1}{2}[\epsilon_a + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}]$	$\frac{1}{3}[\epsilon_a + \epsilon_b + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}]$
Center of Mohr Circle $\frac{s_1 + s_2}{2}$	$\frac{E(\epsilon_a + \epsilon_c)}{2(1 - v)}$	$\frac{E(\epsilon_a + \epsilon_b + \epsilon_c)}{3(1 - v)}$
Maximum Shear Stress τ_{max}	$\frac{E}{2(1 + v)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}$	$\frac{E}{3(1 + v)} \times \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}$
Orientation of Principal Stresses △	$\frac{1}{2}\tan^{-1}\left[\frac{2\epsilon_b - \epsilon_a - \epsilon_c}{\epsilon_a - \epsilon_c}\right]$	$\frac{1}{2}\tan^{-1}\left[\frac{\sqrt{3}\ (\epsilon_c - \epsilon_b)}{(2\epsilon_a - \epsilon_b - \epsilon_c)}\right]$



Stress State

Principal Stresses

$$\tau_{max} = \left[\left(\frac{S_x - S_y}{2}\right)^2 + \tau_{xy}^2\right]^{\frac{1}{2}}$$

$$S_1 = \frac{S_x + S_y}{2} + \tau_{max}$$

$$S_2 = \frac{S_x + S_y}{2} - \tau_{max}$$

$$\theta = \frac{1}{2}\tan^{-1}\left(\frac{2\tau_{xy}}{S_x - S_y}\right)$$

$$S = \frac{S_1 + S_2}{2} + \tau_{max}\cos 2\theta'$$

$$\tau = \tau_{max}\sin 2\theta'$$

Program Listing.

```
I01 LBL I
I02 INPUT E
I03 INPUT V
I04 INPUT A
I05 INPUT B
I06 INPUT C
I07 RCL E
I08 1
I09 RCL- V
I10 ÷
I11 STO J
I12 RCL E
I13 1
I14 RCL+ V
I15 ÷
I16 STO R
I17 RCL A
I18 RCL B
I19 RCL C
I20 RTN
Checksum = 9144
R01 LBL R
R02 XEQ I
R03 RCL+ A
R04 2
R05 ÷
R06 RCL- B
R07 RCL C
R08 RCL A
R09 GTO M
Checksum = FFAA
E01 LBL E
E02 XEQ I
E03 -
E04 3
E05 SQRT
E06 ÷
E07 RCL B
E08 RCL+ C
E09 2
E10 ×
E11 RCL- A
E12 3
E13 ÷
E14 RCL A
E15 GTO M
Checksum = 8A60
S01 LBL S
S02 INPUT S
S03 INPUT Y
S04 INPUT X
S05 1
S06 STO J
S07 STO R
S08 RCL S
S09 +/-
S10 RCL Y
S11 RCL X
Checksum = F899
M01 LBL M
M02 STO L
M03 +
M04 2
M05 ÷
M06 STO× J
M07 RCL- L
M08 ABS
M09 y,x→θ,r
M10 STO× R
M11 x<>y
M12 2
M13 ÷
M14 STO G
M15 RCL R
```

```
M16 RCL+ J              M29 SIN
M17 STO U               M30 LASTx
M18 VIEW U              M31 COS
M19 RCL J               M32 RCL× R
M20 RCL- R              M33 RCL+ J
M21 STO L               M34 STO P
M22 VIEW L              M35 VIEW P
M23 VIEW R              M36 R↓
M24 VIEW G              M37 RCL× R
M25 INPUT W             M38 STO T
M26 RCL+ G              M39 VIEW T
M27 2                   M40 RTN
M28 ×                   Checksum = D351
```

Flags Used. None.

Memory Required. 142.5 bytes.

Remarks.

- Tensile forces are considered positive, compressive stresses negative.
- This program calculates the principal stresses for a two dimensional stress state only. A knowledge of the stresses in the *z*-direction is necessary to determine the overall maximum and minimum stresses.
- Angles must be consistent with the angular mode currently set in the calculator.

Program Instructions.

1. Key in the program listings of the routines to be used; press [C] when finished.
2. Select the appropriate routine:
   - Press [XEQ] E if equiangular strain gage readings are known.
   - Press [XEQ] R if rectangular strain gage readings are known.
   - Press [XEQ] S if stresses are known directly.
3. Key in the variables at each prompt and press [R/S] .
4. See each result as it's displayed. Press [R/S] to display the next one.
5. Optional: At the prompt, key in rotation angle *W* and press [R/S] to obtain the normal stress at that orientation; press [R/S] again to see the shear stress.

Variables Used.

$A = \varepsilon_0$.

$B = \varepsilon_{45}$ or ε_{60}.

$C = \varepsilon_{90}$ or ε_{120}.

E = Young's modulus.

V = Poisson's ratio.

X = normal stress on the x-face, σ_x.

Y = normal stress on the y-face, σ_y.

S = shear stress, τ_{xy}.

U = maximum principal stress, σ_1.

L = minimum principal stress, σ_2.

R = maximum shear stress, τ_{max}.

G = clockwise angle from the specified x-axis to the maximum principal axis.

W = arbitrary angle counterclockwise from the specified x-axis, β.

P = normal stress at angle β.

T = shear stress at angle β.

J = variable used for intermediate results.

Example 1: Equiangular Strain Gage. An equiangular rosette strain gage measures the following strains:

$\varepsilon_0 = 180\ \mu.$
$\varepsilon_{60} = 200\ \mu.$
$\varepsilon_{120} = -290\ \mu.$

Find the principal stresses and their orientation. The material properties are $E = 30 \times 10^6$ psi and $\nu = 0.3$.

Keys:	**Display:**	**Description:**
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] E	E?*value*	Begins equiangular rosette routine.
30 [E] 6 [R/S]	V?*value*	Inputs strain gage readings.
.3 [R/S]	A?*value*	
180 [E] 6 [+/-] [R/S]	B?*value*	
200 [E] 6 [+/-] [R/S]	C?*value*	
290 [+/-] [E] 6 [+/-]		
[R/S]	U=8,675.1358	Displays σ_1.
[R/S]	L=−6,103.7072	Displays σ_2.
[R/S]	R=7,389.4215	Displays τ_{max}.
[R/S]	G=31.0333	Displays θ.

Example 2: Known Stresses. The stresses acting on an element are shown below (all stresses are in MPa).



Find the principal stresses and their orientation, and the stresses on the face of the element oriented 45° counterclockwise from the x-axis.

Keys:	Display:	Description:
[XEQ] S	S?*value*	Begins stress routine.
30 [R/S]	Y?*value*	Inputs stresses.
20 [+/-] [R/S]	X?*value*	
75 [R/S]	U=83.6805	Displays σ_1 .
[R/S]	L= −28.6805	Displays σ_2 .
[R/S]	R=56.1805	Displays τ_{max} .
[R/S]	G= −16.1378	Displays θ .
[R/S]	W?*value*	Inputs angle.
45 [R/S]	P=57.5000	Displays σ.
[R/S]	T=47.5000	Displays τ.

Field Angle Traverse

This program calculates the coordinates of a traverse, the total horizontal distance traversed, and the enclosed area (for a closed traverse). The user must input the northing and easting of the starting point, the reference azimuth, and the direction and distance from each point in the traverse to the next point. The direction may be input either as a deflection right or left or as an angle right or left. The distance may be input either as a horizontal distance or as a slope distance with a zenith angle.

$$HD = SD \sin(ZA)$$

$$N_{k+1} = N_k + HD \cos(AZ)$$

$$E_{k+1} = E_k + HD \sin(AZ)$$

$$LAT_k = N_{k+1} - N_k$$

$$DEP_k = E_{k+1} - E_k$$

$$\text{Area} = \sum_{k=1}^{n} LAT_k \left(\frac{1}{2} DEP_k + \sum_{j=1}^{k-1} DEP_j \right)$$

where:

N, E = northing, easting of a point.

k = a current point.

n = number of points in the survey.

AZ = azimuth of a course.

HD = horizontal distance.

SD = slope distance.

ZA = zenith angle.

Program Listing.

```
F01 LBL F
F02 SF 0
F03 CLVARS
F04 INPUT N
F05 INPUT E
F06 INPUT F
F07 →HR
F08 180
F09 +
F10 STO F
F11 STOP
Checksum = 1CC1
A01 LBL A
A02 INPUT A
A03 →HR
A04 180
A05 +
A06 STO+ F
A07 STOP
Checksum = D137
D01 LBL D
D02 INPUT D
D03 →HR
D04 STO+ F
D05 STOP
Checksum = 025F
S01 LBL S
S02 INPUT S
S03 INPUT Z
S04 →HR
S05 SIN
S06 ×
S07 STO H
S08 CF 0
Checksum = 53D5
H01 LBL H
H02 FS? 0
H03 INPUT H
H04 STO+ T
H05 RCL F
H06 SF 0
H07 x<>y
H08 θ,r→y,x
H09 STO+ N
H10 x<>y
H11 STO+ E
H12 STO X
H13 2
H14 ÷
H15 RCL+ K
H16 ×
H17 STO+ R
H18 RCL X
H19 STO+ K
H20 VIEW N
H21 VIEW E
H22 VIEW T
H23 VIEW R
H24 RTN
Checksum = B55D
```

Flags Used. Flag 0.

Memory Required. 98.5 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- Right angles and deflections are positive; left angles and deflections are negative.
- This program requires the calculator to be set to *degrees* mode; angular inputs must be in degrees-minutes-seconds (D.MS) format.
- The program uses zenith angles to calculate the horizontal distance from slope distance. If you are using vertical angles rather than zenith angles, convert the vertical angle to a zenith angle by using:

  $$\text{zenith angle} = 90° - \text{vertical angle.}$$

  (Remember to convert D.MS input to decimal degrees before subtracting from 90.)

Program Instructions.

1. Key in the program listing and press [C] when finished.
2. Press [XEQ] F.
3. Key in N and press [R/S] ; key in E and press [R/S] ; key in F and press [R/S] .
4. Select the appropriate routine to input the direction:
   - For an angle right or an angle left:
     - Press [XEQ] A.
     - Key in *A*.
     - If angle left, press [+/−] .
     - Press [R/S] .
   - For a deflection right or a deflection left:
     - Press [XEQ] D.
     - Key in *D*.
     - If deflection left, press [+/−] .
     - Press [R/S] .

5. Select the appropriate routine to input the distance:
   - For a horizontal distance:
     - Press [XEQ] H.
     - Key in A and press [R/S] .
     - See N and press [R/S] .
     - See E.
   - For a slope distance with zenith angle:
     - Press [XEQ] S.
     - Key in S and press [R/S] .
     - Key in Z and press [R/S] .
     - See N and press [R/S] .
     - See E.
6. When the final distance and angle have been keyed in, press [R/S] to see T.
7. Press [R/S] to see R.

Variables Used.

N = the northing of each point.

E = the easting of each point.

F = reference azimuth away from the starting point.

A = angle change of direction.

D = deflection change of direction.

H = horizontal distance.

S = slope distance.

Z = zenith angle.

T = total distance traversed.

R = enclosed area.

X, K = variables used for intermediate results.

Example: Field Angle Traverse. Find the coordinates of each point, the total distance, and the area enclosed for the following field:



Keys:	Display:	Description:
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] F	N?0.0000	
150 [R/S]	E?0.0000	Inputs starting point data.
400 [R/S]	F?0.0000	
311.3955 [R/S]	491.6653	
[XEQ] A	A?0.0000	
113.3455 [R/S]	293.5819	
[XEQ] H	H?0.0000	
177.966 [R/S]	N=224.5150	Displays N_2.
[R/S]	E=561.6150	Displays E_2.
[XEQ] D	D?0.0000	
100.2455 [+/-] [R/S]	-100.4153	
[XEQ] S	S?0.0000	
161.88 [R/S]	Z?0.000	
86.0139 [R/S]	N=356.5285	Displays N_3.
[R/S]	E=468.5999	Displays E_3.
[XEQ] A	A?113.3455	
87.3559 [R/S]	267.5997	
[XEQ] H	H?161.4911	
203.69 [R/S]	N=232.3373	Displays N_4.
[R/S]	E=307.1498	Displays E_4.
[XEQ] D	D?-100.2455	
100.4559 [+/-] [R/S]	-100.7664	
[XEQ] H	H?203.6900	
124 [R/S]	N=149.9048	Displays N_1.
[R/S]	E=399.7829	Displays E_1.
[R/S]	T=667.1471	Displays total distance.
[R/S]	R=26,558.8326	Displays area.

4

Statistics

t Statistics

This program performs t statistics calculations for either paired statistics or for two means.

Paired t Statistics

Given a set of paired observations from two normal populations with unknown means μ_1 and μ_2:

x_i	x_1	$x_2 \cdots x_n$
y_i	y_1	$y_2 \cdots y_n$

let:

$$D_i = x_i - y_i$$

$$\overline{D} = \frac{1}{n}\sum_{i=1}^{n} D_i$$

$$s_D = \left[\frac{\Sigma D_i^2 - \frac{1}{n}(\Sigma D_i)^2}{n - 1}\right]^{\frac{1}{2}}$$

The test statistic

$$t = \frac{\overline{D}}{s_D}\sqrt{n}$$

has $n - 1$ degrees of freedom (df) and can be used to test the null hypothesis $H_0 : \mu_1 = \mu_2$.

t Statistic for Two Means

Suppose $\{x_1, x_2, ..., x_{n_1}\}$ and $\{y_1, y_2, ..., y_{n_2}\}$ are independent random samples from two normal populations having means μ_1 and μ_2 (unknown) and the same unknown variance σ^2.

To test the null hypothesis $H_0 : \mu_1 - \mu_2 = d$, where d is a given number, use the following equations:

$$\overline{x} = \frac{1}{n_1}\sum_{i=1}^{n_1} x_i$$

$$\overline{y} = \frac{1}{n_2}\sum_{i=1}^{n_2} y_i$$

$$t = \frac{\overline{x} - \overline{y} - d}{\left[\left(\frac{1}{n_1} + \frac{1}{n_2}\right)\left(\frac{\Sigma x_i^2 - n_1\overline{x}^{\,2} + \Sigma y_i^2 - n_2\overline{y}^{\,2}}{n_1 + n_2 - 2}\right)\right]^{\frac{1}{2}}}$$

You can use this t statistic, which has the t distribution with $n_1 + n_2 - 2$ degrees of freedom (*df*), to test the null hypothesis H_0.

Program Listing.

```
P01 LBL P
P02 CLΣ
P03 INPUT N
Checksum = FCFE
K01 LBL K
K02 INPUT X
K03 INPUT Y
K04 RCL X
K05 RCL- Y
K06 Σ+
K07 DSE N
K08 GTO K
K09 x̄
K10 STO D
K11 sx
K12 STO S
K13 ÷
K14 n
K15 SQRT
K16 ×
K17 STO T
K18 n
K19 1
K20 -
K21 STO F
K22 VIEW D
K23 VIEW S
K24 VIEW T
K25 VIEW F
K26 RTN
Checksum = 16B7
T01 LBL T
T02 INPUT H
T03 CLΣ
T04 STOP
T05 x̄
T06 STO A
T07 Σx²
T08 STO B
T09 n
T10 STO C
T11 CLΣ
T12 STOP
T13 x̄
T14 STO J
T15 Σx²
T16 STO K
T17 n
T18 STO L
T19 RCL B
T20 RCL A
T21 x²
T22 RCL× C
T23 -
T24 RCL+ K
T25 RCL J
T26 x²
T27 RCL× L
T28 -
T29 RCL C
T30 RCL+ L
T31 2
T32 -
T33 STO F
T34 ÷
T35 SQRT
T36 1/x
T37 RCL A
T38 RCL- J
T39 RCL- H
T40 ×
T41 RCL C
T42 1/x
T43 RCL L
```

```
T44 1/x                T49 VIEW T
T45 +                  T50 VIEW F
T46 SQRT               T51 RTN
T47 ÷                  Checksum = A79C
T48 STO T
```

Flags Used. None.

Memory Required. 120 bytes.

Remarks.

- *This program clears all statistical data stored in Continuous Memory.*
- The two routines are independent of each other.

Program Instructions.

1. Key in the programs to be used; press [C] when finished.
2. For paired t statistics:
   - Press [XEQ] P.
   - Key in *N* and press [R/S] .
   - Key in each *X* and press [R/S] ; key in the corresponding *Y* and press [R/S] .
   - See the results as they are displayed. Press [R/S] to display the next result.
3. For t statistics of two means:
   - Press [XEQ] T.
   - Key in *H* and press [R/S] .
   - Key in each *x*-value and press [Σ+] .
   - When all of the *x*-values have been entered, press [R/S] .
   - Key in each *y*-value and press [Σ+] .
   - When all of the *y*-values have been entered, press [R/S] to calculate the test statistic *T*.
   - Press [R/S] to calculate the degrees of freedom.

Variables Used.

X = x-value of a pair of observations.

Y = y-value of a pair of observations.

N = number of paired values.

D = average difference, $\overline{D}$.

S = standard deviation, s_D.

F = number of degrees of freedom, df.

H = null hypothesis difference, d.

T = test statistic.

A, B, C, J, K, L = variables used for intermediate results.

Example 1: Paired Observations. Calculate the test statistic and degrees of freedom of the following data pairs for the null hypothesis $H_0 : \mu_1 = \mu_2$.

x	15	16.9	15.3	17	19.1	15.3
y	18	19.3	17	20.3	19.7	18

Keys:	**Display:**	**Description:**
[XEQ] P	N?*value*	
6 [R/S]	X?*value*	Inputs values.
15 [R/S]	Y?*value*	
18 [R/S]	X?15.0000	
16.9 [R/S]	Y?18.0000	
19.3 [R/S]	X?16.9000	
15.3 [R/S]	Y?19.3000	
17 [R/S]	X?15.3000	
17 [R/S]	Y?17.0000	
20.3 [R/S]	X?17.0000	
19.1 [R/S]	Y?20.3000	
19.7 [R/S]	X?19.1000	

15.3 [R/S]	Y?19.7000	
18 [R/S]	D= −2.2833	Displays $\overline{D}$.
[R/S]	S=0.9908	Displays s_D.
[R/S]	T= −5.6450	Displays t.
[R/S]	F=5.0000	Displays df .

Example 2: Two Means. Calculate the test statistic and degrees of freedom of the following data for the null hypothesis $H_0 : \mu_1 = \mu_2$.

x	86	109	112	91	103	121	107	100	97
y	93	101	111	117	105	97	99		

Keys:	**Display:**	**Description:**
[XEQ] T	H?*value*	
0 [R/S]	0.0000	
86 [Σ+]	1.0000	Inputs x -values.
109 [Σ+]	2.0000	
112 [Σ+]	3.0000	
91 [Σ+]	4.0000	
103 [Σ+]	5.0000	
121 [Σ+]	6.0000	
107 [Σ+]	7.0000	
100 [Σ+]	8.0000	
97 [Σ+]	9.0000	
[R/S]	9.0000	
93 [Σ+]	1.0000	Inputs y -values.
101 [Σ+]	2.0000	
111 [Σ+]	3.0000	
117 [Σ+]	4.0000	
105 [Σ+]	5.0000	
97 [Σ+]	6.0000	
99 [Σ+]	7.0000	
[R/S]	T= −0.0801	Displays t.
[R/S]	F=14.0000	Displays df.

Chi-Square Evaluation

This program calculates the value of the χ^2 statistic for the goodness of fit test using the equation:

$$\chi^2 = \sum_{i=1}^{n} \frac{(O_i - E_i)^2}{E_i} \qquad (df = n - 1)$$

where:

O_i = the observed frequency.

E_i = the expected frequency.

If the expected values are equal:

$$E = E_i = \frac{\Sigma O_i}{n} \quad \text{for all } i$$

then:

$$\chi^2 = \frac{n \, \Sigma O_i^2}{\Sigma O_i} - \Sigma O_i$$

Program Listing.

```
U01 LBL U
U02 0
U03 STO C
U04 INPUT N
Checksum = 05C2
D01 LBL D
D02 INPUT O
D03 INPUT E
D04 RCL O
D05 RCL- E
D06 x²
D07 RCL÷ E
D08 STO+ C
D09 DSE N
D10 GTO D
D11 VIEW C
D12 RTN
Checksum = 989C
E01 LBL E
E02 CLΣ
E03 INPUT N
Checksum = 222E
K01 LBL K
K02 INPUT O
K03 Σ+
K04 DSE N
K05 GTO K
K06 n
K07 Σx²
K08 x
```

```
K09 Σx
K10 ÷
K11 LASTx
K12 -
K13 STO C
K14 VIEW C
K15 x̄
K16 STO E
K17 VIEW E
K18 RTN
Checksum = 23FF
```

Flags Used. None.

Memory Required. 55.5 bytes.

Remarks.

- *This program clears all statistical data stored in Continuous Memory.*
- The two routines (the unequal case and the equal case) are independent of each other.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Select the appropriate routine:
   - Press [XEQ] U if the expected values are unequal.
   - Press [XEQ] E if the expected values are equal.
3. Key in the variables at each prompt and press [R/S] .
4. After all data is input, the χ^2 value is calculated and displayed.
5. For a new case, go to step 2.

Variables Used.

O = observed frequency.

E = expected frequency.

$C = \chi^2$.

N = number of data pairs (unequal case) or values (equal case).

Example 1: Unequal Expected Frequencies. Find the χ^2 statistic for the goodness of fit for the following data set:

Observed	8	50	47	56	5	14
Expected	9.6	46.75	51.85	54.4	8.25	9.15

Keys:	**Display:**	**Description:**
[XEQ] U	N?*value*	
6 [R/S]	O?*value*	Inputs values.
8 [R/S]	E?*value*	
9.6 [R/S]	O?8.0000	
50 [R/S]	E?9.6000	
46.75 [R/S]	O?50.0000	
47 [R/S]	E?46.7500	
51.85 [R/S]	O?47.0000	
56 [R/S]	E?51.8500	
54.4 [R/S]	O?56.0000	
5 [R/S]	E?54.4000	
8.25 [R/S]	O?5.0000	
14 [R/S]	E?8.2500	
9.15 [R/S]	C=4.8444	Displays χ^2.

Example 2: Equal Expected Frequencies. The following table shows the frequencies observed in tossing a die 120 times. χ^2 can be used to test if the die is fair ($df = 5$). Assume that the expected frequencies are equal.

i	1	2	3	4	5	6
Observed	25	17	15	23	24	16

Keys:	**Display:**	**Description:**
[XEQ] E	N?*value*	
6 [R/S]	O?*value*	Inputs values.
25 [R/S]	O?25.0000	
17 [R/S]	O?17.0000	
15 [R/S]	O?15.0000	
23 [R/S]	O?23.0000	
24 [R/S]	O?24.0000	
16 [R/S]	C=5.0000	Calculates and displays χ^2.

The value of χ^2 for $df = 5$ and 5% significance* is 11.070. Since 5.00 is less than 11.070, no statistically significant differences exist between the observed and expected frequencies.

* See page 438 of J.E. Freund's *Mathematical Statistics*, 2nd edition.

F Distribution

This program evaluates the integral of the F distribution:

$$Q(x) = \int_x^\infty \left[\frac{\Gamma\left(\frac{\nu_1 + \nu_2}{2}\right) y^{\frac{\nu_1}{2} - 1} \left(\frac{\nu_1}{\nu_2}\right)^{\frac{\nu_1}{2}}}{\Gamma\left(\frac{\nu_1}{2}\right)\Gamma\left(\frac{\nu_2}{2}\right)\left(1 + \frac{\nu_1}{\nu_2} y\right)^{\frac{\nu_1 + \nu_2}{2}}} \right] dy$$

where x > 0 and ν_1 and ν_2 are the degrees of freedom, provided either ν_1 or ν_2 is even and both are greater than two.



The following series are used to evaluate the integral.

If ν_1 is even:

$$Q(x) = t^{\frac{\nu_2}{2}} \left[1 + \frac{\nu_2}{2}(1-t) + \frac{\nu_2(\nu_2+2)}{2 \cdot 4}(1-t)^2 + \cdots + \frac{\nu_2(\nu_2+2)\cdots(\nu_2+\nu_1-4)}{2 \cdot 4 \cdots (\nu_1-2)}(1-t)^{\frac{\nu_1-2}{2}} \right]$$

If ν_2 is even:

$$Q(x) = 1 - (1-t)^{\frac{\nu_1}{2}} \left[1 + \frac{\nu_1}{2} t + \frac{\nu_1(\nu_1+2)}{2 \cdot 4} t^2 + \cdots + \frac{\nu_1(\nu_1+2)\cdots(\nu_2+\nu_1-4)}{2 \cdot 4 \cdots (\nu_2-2)} t^{\frac{\nu_2-2}{2}} \right]$$

where:

$$t = \frac{\nu_2}{\nu_2 + \nu_1 x}$$

Program Listing.

```
F01 LBL F
F02 CF 0
F03 INPUT A
F04 INPUT B
F05 INPUT X
F06 RCL× A
F07 RCL+ B
F08 RCL÷ B
F09 1/x
F10 STO T
F11 RTN
Checksum = 4D69
E01 LBL E
E02 SF 0
E03 RCL A
E04 RCL B
E05 STO A
E06 x<>y
E07 STO B
E08 1
E09 RCL- T
E10 STO T
Checksum = 2FFE
D01 LBL D
D02 0
D03 STO I
```

```
D04 1
D05 STO K
D06 STO H
D07 RCL A
D08 2
D09 -
D10 2
D11 ÷
D12 STO N
Checksum = 9A01
P01 LBL P
P02 1
P03 STO+ I
P04 RCL- T
P05 RCL× K
P06 2
P07 RCL I
P08 y^x
P09 ÷
P10 RCL I
P11 1
P12 -
P13 2
P14 ×
P15 RCL+ B
P16 ×
P17 STO K
P18 STO+ H
P19 RCL N
P20 RCL- I
P21 x≠0?
P22 GTO P
P23 RCL B
P24 2
P25 ÷
P26 RCL T
P27 x<>y
P28 y^x
P29 RCL× H
P30 1
P31 x<>y
P32 FS? 0
P33 -
P34 STO Q
P35 VIEW Q
P36 RTN
Checksum = 5194
```

Flags Used. Flag 0.

Memory Required. 103.5 bytes.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] F.
3. Key in the variables as they are prompted for, pressing [R/S] after each entry.
4. Select the appropriate routine to calculate $Q(x)$:
   - Press [XEQ] E if ν_1 is odd and ν_2 is even.
   - Press [XEQ] D if ν_1 is even and ν_2 is odd.
   - Press [XEQ] D if both ν_1 and ν_2 are even.
5. For a new case, go to step 2.

Variables Used.

$A = \nu_1$.

$B = \nu_2$.

$X = x$.

$Q = Q(x)$.

T, I, K, H, N = variables used for intermediate results.

Example 1. Calculate $Q(3.92)$, where $\nu_1 = 9$ and $\nu_2 = 6$.

Keys:	Display:	Description:
[XEQ] F	A?*value*	
9 [R/S]	B?*value*	Inputs values.
6 [R/S]	X?*value*	
3.92 [R/S]	0.1453	
[XEQ] E	Q=0.0552	Displays $Q(3.92)$.

Example 2. Calculate $Q(1.85)$, where $\nu_1 = 4$ and $\nu_2 = 16$.

Keys:	Display:	Description:
[XEQ] F	A?*value*	
4 [R/S]	B?*value*	Inputs values.
16 [R/S]	X?*value*	
1.85 [R/S]	0.6838	
[XEQ] D	Q=0.1687	Displays $Q(1.85)$.

Analysis of Variance (One Way)

One way analysis of variance tests the difference between the population means of k treatment groups. Group i $(i = 1, 2, \ldots, k)$ has n_i observations. Treatment groups may have equal or unequal numbers of observations.

$$\text{Sum}_i = \sum \text{ of observations in treatment group } i = \sum_{j=1}^{n_i} x_{ij}$$

$$\text{Total } SS = \sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij}^2 - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij} \right)^2}{\sum_{i=1}^{k} n_i}$$

$$\text{Treat } SS = \sum_{i=1}^{k} \left[\frac{\left(\sum_{j=1}^{n_i} x_{ij} \right)^2}{n_i} \right] - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} x_{ij} \right)^2}{\sum_{i=1}^{k} n_i}$$

$$\text{Error } SS = \text{Total } SS - \text{Treat } SS$$

$$df_1 = \text{Treat } df = k - 1$$

$$df_2 = \text{Error } df = \sum_{i=1}^{k} n_i - k$$

$$\text{Treat } MS = \frac{\text{Treat } SS}{\text{Treat } df}$$

$$\text{Error } MS = \frac{\text{Error } SS}{\text{Error } df}$$

$$F = \frac{\text{Treat } MS}{\text{Error } MS} \quad \left(\text{with } k - 1 \text{ and } \sum_{i=1}^{k} n_i - k \text{ degrees of freedom}\right)$$

Program Listing.

```
A01 LBL A
A02 CLVARS
A03 INPUT K
A04 STO J
Checksum = B956
L01 LBL L
L02 CLx
L03 CLΣ
L04 STOP
L05 Σx
L06 STO S
L07 VIEW S
L08 STO+ A
L09 Σx²
L10 STO+ B
L11 n
L12 STO+ N
L13 Σx
L14 x²
L15 n
L16 ÷
L17 STO+ C
L18 DSE J
L19 GTO L
L20 RCL B
L21 RCL A
L22 x²
L23 RCL÷ N
L24 -
L25 LASTx
L26 +/-
L27 RCL+ C
L28 -
L29 LASTx
L30 ÷
L31 1/x
L32 RCL K
L33 1
L34 -
L35 STO D
L36 VIEW D
L37 ÷
L38 RCL N
L39 RCL- K
L40 STO D
L41 VIEW D
L42 ×
L43 STO F
L44 VIEW F
L45 RTN
Checksum = 0443
```

Flags Used. None.

Memory Required. 73.5 bytes.

Remarks. *This program clears all variables and statistical data stored in Continuous Memory.*

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] A.
3. Key in K (the number of treatment groups) and press [R/S].
4. Key in each observation and press [Σ+] .
5. Press [R/S] when all of the observations in the treatment group have been entered.
6. See the sum and press [R/S] .
7. See the treatment degrees of freedom (df_1) and press [R/S] .
8. See the error degrees of freedom (df_2) and press [R/S] .
9. See the F ratio (F).

Variables Used.

K = number of treatment groups.

S = sum of observations in a treatment group.

D = df_1 and df_2 .

F = F ratio.

J, A, B, N, C = variables used for intermediate results.

Example. Find Sum_1, Sum_2, Sum_3, df_1, df_2, and F for the following:

i \ j	1	2	3	4	5	6
1	10	13	12	15	17	
2	8	10	12	13	11	9
3	8	9	12	7		

Keys:	**Display:**	**Description:**
[XEQ] A	K?0.0000	Prompts for number of treatment groups.
3 [R/S]	0.0000	
10 [Σ+]	1.0000	Inputs observations in first treatment group.
13 [Σ+]	2.0000	
12 [Σ+]	3.0000	
15 [Σ+]	4.0000	
17 [Σ+]	5.0000	
[R/S]	S=67.0000	Displays Sum_1.
[R/S]	0.0000	Inputs observations in second treatment group.
8 [Σ+]	1.0000	
10 [Σ+]	2.0000	
12 [Σ+]	3.0000	
13 [Σ+]	4.0000	
11 [Σ+]	5.0000	
9 [Σ+]	6.0000	
[R/S]	S=63.000	Displays Sum_2.
[R/S]	0.0000	Inputs observations in third observation group.
8 [Σ+]	1.0000	
9 [Σ+]	2.0000	
12 [Σ+]	3.0000	
7 [Σ+]	4.0000	
[R/S]	S=36.0000	Displays Sum_3.
[R/S]	D=2.0000	Displays df_1.
[R/S]	D=12.0000	Displays df_2.
[R/S]	F=4.5700	Displays F.

Binomial Distribution

This program calculates the probability of a value falling within a specified range of values, that is, the cummulative distribution $\sum_{x=B}^{A} p(x)$, in a binomial distribution. It also calculates the mean, the variance, and the standard deviation of the distribution, and can be used to find the value of each term in the distribution.

$$p(x) = \binom{n}{x} r^x (1 - r)^{n-x}$$

where $x = 0, 1, 2, \ldots$ and $r < 1$.

Program Listing.

```
B01 LBL B
B02 CLVARS
B03 INPUT N
B04 INPUT R
B05 INPUT B
B06 INPUT A
B07 1,000
B08 ÷
B09 STO+ B
Checksum = 6A51
Y01 LBL Y
Y02 1
Y03 RCL- R
Y04 RCL N
Y05 RCL B
Y06 IP
Y07 -
Y08 y^x
Y09 RCL R
Y10 RCL B
Y11 IP
Y12 y^x
Y13 LASTx
Y14 RCL N
Y15 x<>y
Y16 Cn,r
Y17 ×
Y18 ×
Y19 FS? 0
Y20 STOP
Y21 STO+ P
Y22 ISG B
Y23 GTO Y
Y24 VIEW P
Y25 RCL N
Y26 RCL× R
Y27 STO M
Y28 1
Y29 RCL- R
Y30 ×
Y31 STO V
Y32 SQRT
```

```
Y33 STO S          Y36 VIEW S
Y34 VIEW M         Y37 RTN
Y35 VIEW V         Checksum = 2C13
```

Flags Used. Flag 0.

Memory Required. 77 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- The upper and lower limits of the range are inclusive ($B \leq x \leq A$). If the limits are exclusive or noninteger values, round the lower limit to the next highest integer and the upper limit to the next lowest integer.
- The limits A and B have no effect on the mean, variance, and standard deviation.
- An invalid data error will result if $B < 0$ or if $A > n$.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] B.
3. Key in the variables at each prompt and press [R/S].
4. Optional: To see each term of the distribution, set flag 0; press [R/S] to continue execution.
5. See each result as it is displayed and press [R/S].
6. For a new case, go to step 2.

Variables Used.

N = number of events.

R = probability of the occurrence of a single event.

A = upper limit of the range.

B = lower limit of the range.

P = probability of a value falling in the range.

M = mean.

V = variance.

S = standard deviation.

Example 1. A fair coin ($r = 0.5$) is tossed 10 times. What is the probability that at least seven heads will occur? Find the mean, variance, and standard deviation.

Keys:	Display:	Description:
■ [FLAGS] {CF} 0	*value*	Clears flag 0.
[XEQ] B	N?0.0000	
10 [R/S] .5 [R/S] 7 [R/S]	R?0.0000 B?0.0000 A?0.0000	Inputs values.
10 [R/S] [R/S] [R/S] [R/S]	P=0.1719 M=5.0000 V=2.5000 S=1.5811	Displays probability. Displays mean. Displays variance. Displays standard deviation.

Example 2. Find the terms of the binomial distribution with $n = 5$ and $r = 0.75$.

Keys:	Display:	Description:
■ [FLAGS] {SF} 0	*value*	Sets flag to display each term of distribution.
[XEQ] B	N?0.0000	
5 [R/S]	R?0.0000	Inputs values.
.75 [R/S]	B?0.0000	
[R/S]	A?0.0000	
5 [R/S]	0.0010	Displays p (0).
[R/S]	0.0146	Displays p (1).
[R/S]	0.0879	Displays p (2).
[R/S]	0.2637	Displays p (3).
[R/S]	0.3955	Displays p (4).
[R/S]	0.2373	Displays p (5).

Poisson Distribution

This program calculates the probability of a value falling within a specified range of values, that is, the cummulative distribution $\sum_{x=B}^{A} p(x)$, in a Poisson distribution. It also calculates the mean, the variance, and the standard deviation of the distribution, and can be used to find the value of each term in the distribution.

$$p(x) = \frac{e^{-\lambda}\lambda^{x}}{x!}$$

where $x = 0, 1, 2, \ldots$ and $\lambda > 0$

Program Listing.

```
P01 LBL P
P02 CLVARS
P03 INPUT L
P04 INPUT B
P05 INPUT A
P06 1,000
P07 ÷
P08 STO+ B
Checksum = 0BD9
X01 LBL X
X02 RCL L
X03 RCL B
X04 IP
X05 y^x
X06 LASTx
X07 x!
X08 ÷
X09 RCL L
X10 +/-
X11 e^x
X12 x
X13 FS? 0
X14 STOP
X15 STO+ P
X16 ISG B
X17 GTO X
X18 VIEW P
X19 RCL L
X20 STO M
X21 STO V
X22 SQRT
X23 STO S
X24 VIEW M
X25 VIEW V
X26 VIEW S
X27 RTN
Checksum = B9BA
```

Flags Used. Flag 0.

Memory Required. 60.5 bytes.

Remarks.

- *This program clears all variables stored in Continuous Memory.*
- The upper and lower limits of the range are inclusive ($B \leq x \leq A$). If the limits are exclusive or noninteger values, round the lower limit to the next highest integer and the upper limit to the next lowest integer.
- The limits A and B have no effect on the mean, variance, and standard deviation. A must be ≤ 999, B must be ≥ 0.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] P.
3. Key in the variables at each prompt and press [R/S] .
4. Optional: To see each term of the distribution, set flag 0; press [R/S] to continue execution.
5. See each result as it is displayed and press [R/S] .
6. For a new case, go to step 2.

Variables Used.

$L = \lambda$.

A = upper limit of the range.

B = lower limit of the range.

P = probability of a value falling in the range.

M = mean.

V = variance.

S = standard deviation.

Example 1. For a Poisson distribution with $\lambda = 2$, find the probability that $0 < x < 2.5$; also find the mean, the variance, and the standard deviation. (Remember that the Poisson distribution deals only with integers. Therefore, the only x values in this range are 1 and 2.)

Keys:	Display:	Description:
■ [FLAGS] {CF} 0	*value*	Clears flag 0.
[XEQ] P	L?0.0000	
2 [R/S]	B?0.0000	Inputs values.
1 [R/S]	A?0.0000	
2 [R/S]	P=0.5413	Displays probability.
[R/S]	M=2.0000	Displays mean.
[R/S]	V=2.0000	Displays variance.
[R/S]	S=1.4142	Displays standard deviation.

Example 2. Find the first six terms (from $x = 0$ to $x = 5$) of the Poisson distribution with $\lambda = 3$.

Keys:	Display:	Description:
■ [FLAGS] {SF} 0	*value*	Sets flag to display each term of distribution.
[XEQ] P	L?0.0000	
3 [R/S]	B?0.0000	Inputs values.
[R/S]	A?0.0000	
5 [R/S]	0.0498	Displays p (0).
[R/S]	0.1494	Displays p (1).
[R/S]	0.2240	Displays p (2).
[R/S]	0.2240	Displays p (3).
[R/S]	0.1680	Displays p (4).
[R/S]	0.1008	Displays p (5).

5

Mathematics

Triangle Solutions

This program may be used to find the sides, angles, and area of a plane triangle.



In general, the specifications of any three of the six parameters of a triangle (three sides and three angles) is sufficient to define a triangle (the exception is that three angles will not define a triangle).
This program will handle all five cases:

- Three sides (SSS).
- Two angles and the included side (ASA).
- Two angles and the adjacent side (SAA).
- Two sides and the included angle (SAS).
- Two sides and the adjacent angle (SSA).

The last case listed (SSA) may result in two solutions to the triangle. This program will calculate both solutions.

If the three known input values are selected in a clockwise order around the triangle, the output values will also follow a clockwise order.

Program Listing.

```
A01 LBL A
A02 INPUT A
A03 INPUT C
A04 INPUT E
A05 RCL A
A06 RCL C
A07 y,x→θ,r
A08 x²
A09 RCL E
A10 x²
A11 -
A12 2
A13 RCL× A
A14 RCL× C
A15 ÷
A16 ACOS
A17 STO B
A18 GTO K
Checksum = 9E72
B01 LBL B
B02 INPUT F
B03 INPUT A
B04 INPUT B
B05 RCL F
B06 SIN
B07 RCL B
B08 RCL+ F
B09 SIN
B10 ÷
B11 RCL× A
B12 STO C
B13 GTO K
Checksum = 5DAB
C01 LBL C
C02 INPUT A
C03 INPUT B
C04 INPUT D
C05 RCL+ B
C06 SIN
C07 RCL D
C08 SIN
C09 ÷
C10 RCL× A
C11 STO C
C12 GTO K
Checksum = D18A
E01 LBL E
E02 CF 2
E03 INPUT A
E04 INPUT C
E05 RCL A
E06 RCL C
E07 x>y?
E08 SF 2
E09 INPUT D
E10 SIN
E11 RCL÷ A
E12 ×
E13 ASIN
E14 RCL+ D
E15 XEQ Z
E16 STO B
E17 XEQ K
E18 RCL F
E19 XEQ Z
E20 STO F
E21 RCL+ D
E22 XEQ Z
E23 STO B
E24 GTO K
Checksum = FB3E
D01 LBL D
D02 INPUT A
D03 INPUT B
```

```
D04 INPUT C
Checksum = 2A67
K01 LBL K
K02 RCL B
K03 RCL A
K04 θ,r→y,x
K05 RCL- C
K06 +/-
K07 y,x→θ,r
K08 STO E
K09 x<>y
K10 STO D
K11 RCL+ B
K12 XEQ Z
K13 STO F
K14 SIN
K15 ×
K16 RCL× A
K17 2
K18 ÷
K19 STO G
K20 VIEW A
K21 VIEW B
K22 VIEW C
K23 VIEW D
K24 VIEW E
K25 VIEW F
K26 VIEW G
K27 RTN
Checksum = 80F0
Z01 LBL Z
Z02 COS
Z03 +/-
Z04 ACOS
Z05 RTN
Checksum = 929B
```

Flags Used. Flag 2.

Memory Required. 154.5 bytes.

Remarks.

- Angles must be consistent with the angular mode currently set in the calculator.
- Routines A through E are independent of each other. Therefore, key in only those routines that will be used. Routines K and Z must be keyed in to use any of the five routines.
- The triangle notation used by this program is *not* consistent with standard triangle notation; in other words, A_1 is not opposite S_1.
- The accuracy of the solution decreases for triangles containing extremely small angles.

Program Instructions.

1. Key in the programs to be used; press [C] when finished.
2. Select the appropriate routine:
   - Press [XEQ] A if three sides are known (SSS).
   - Press [XEQ] B if two angles and an included side are known (ASA).
   - Press [XEQ] C if two angles and an adjacent side are known (SAA).
   - Press [XEQ] D if two sides and an included angle are known (SAS).
   - Press [XEQ] E if two sides and an adjacent angle are known (SSA).
3. Key in the variables at each prompt and press [R/S] .
4. See each result as it is displayed. Press [R/S] for subsequent results.
5. If flag 2 is displayed on the calculator screen while executing routine E, a second possible solution exists. Press [R/S] and return to step 4 to see the second set of results.

Variables Used.

A = side_1.

B = angle_1.

C = side_2.

D = angle_2.

E = side_3.

F = angle_3.

G = the area.

Example 1: Three Known Sides. A farmer uses three sections of straight fence to enclose a field. The lengths are 100 feet, 120 feet, and 150 feet. Find the area enclosed and the angles formed.

Keys:	**Display:**	**Description:**
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] A	A?*value*	
100 [R/S]	C?*value*	Inputs lengths.
120 [R/S]	E?*value*	
150 [R/S]	A=100.0000	Displays S_1.
[R/S]	B=85.4593	Displays A_1.
[R/S]	C=120.0000	Displays S_2.
[R/S]	D=41.6497	Displays A_2.
[R/S]	E=150.0000	Displays S_3.
[R/S]	F=52.8910	Displays A_3.
[R/S]	G=5,981.1684	Displays area.

Example 2: Two Possible Solutions. Given two sides and a nonincluded angle, solve for the triangle.

$$S_1 = 22.5$$
$$S_2 = 37.5$$
$$A_2 = 31.3°$$



Keys:	Display:	Description:
■ [MODES] {DG}		Sets *degrees* mode.
[XEQ] E	A?*value*	
22.5 [R/S]	C?*value*	Inputs values. Flag 2 is displayed.
37.5 [R/S]	D?*value*	
31.3 [R/S]	A=22.5000	Displays S_1.
[R/S]	B=88.7184	Displays A_1 (first solution).
[R/S]	C=37.5000	Displays S_2.
[R/S]	D=31.3000	Displays A_2.
[R/S]	E=43.2984	Displays S_3.
[R/S]	F=59.9816	Displays A_3.
[R/S]	G=421.7695	Displays area.
[R/S]	A=22.5000	Displays S_1. (second soution).
[R/S]	B=28.6816	Displays A_1.
[R/S]	C=37.5000	Displays S_2.
[R/S]	D=31.3000	Displays A_2.
[R/S]	E=20.7860	Displays S_3.
[R/S]	F=120.0184	Displays A_3.
[R/S]	G=202.4757	Displays area.

Derivative of a Function

This program calculates the derivative of a function at a given value. The function must be defined by a separate program label.

$$f'(x) \approx \frac{f(x+\delta) - f(x-\delta)}{2\delta}$$

Program Listing.

```
D01 LBL D
D02 INPUT i
D03 INPUT X
D04 ABS
D05 x≠0?
D06 LOG
D07 IP
D08 4
D09 -
D10 10^x
D11 STO Y
D12 RCL+ X
D13 XEQ(i)
D14 STO Z
D15 RCL X
D16 RCL- Y
D17 XEQ(i)
D18 RCL Z
D19 -
D20 +/-
D21 RCL÷ Y
D22 2
D23 ÷
D24 STO D
D25 VIEW D
D26 RTN
Checksum = 20DF
```

Flags Used. None.

Memory Required. 39 bytes.

Remarks. The program defining the function must place the value of the function in the X-register.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Key in the program that defines the function. The program should take the value in the X-register as input and leave the resulting value of the function in the X-register as output.
3. Press [XEQ] D. When the prompt i?*value* is displayed, specify the function by entering the number between 1 and 26 corresponding to

the program label (in other words, $A = 1$, $B = 2$, and so on), then press [R/S] .

4. Key in the X-value where the function is to be evaluated and press [R/S] to display the derivative at that value.
5. For a new point, go to step 3.
6. For a new function, go to step 2.

Variables Used.

D = derivative of the function.

X = the value at which the derivative of the function is evaluated.

i = the index variable, used to specify the label of the program that defines the function.

Y, Z = variables used for intermediate results.

Example. If $f(x) = 3ln(x^2 - 1)$, find df/dx when $x = 1.5$.

First, enter the program for the function:

```
F01 LBL F
F02 x²
F03 1
F04 -
F05 LN
F06 3
F07 ×
F08 RTN
Checksum = 3F9B
```

Then follow these keystrokes:

Keys:	Display:	Description:
[XEQ] D	i?*value*	Prompts for number corresponding to program label that defines the function.
6 [R/S]	X?*value*	Prompts for value where the derivative of the function is to be evaluated.
1.5 [R/S]	D=7.2000	Displays the derivative of the function.

Linear Interpolation

Numerical relationships are often available in the form of tables. This program uses a straight line approximation to estimate a y-value given a corresponding x-value. Two pairs of x- and y-values of the relationship must be known.

$$y = \left(\frac{y_1 - y_2}{x_1 - x_2}\right)(x - x_1) + y_1$$

Program Listing.

```
L01 LBL L
L02 INPUT X
L03 INPUT A
L04 INPUT B
L05 INPUT C
L06 INPUT D
L07 RCL C
L08 RCL- D
L09 RCL A
L10 RCL- B
L11 ÷
L12 RCL X
L13 RCL- A
L14 ×
L15 RCL+ C
L16 STO Y
L17 VIEW Y
L18 RTN
Checksum = 96CF
```

Flags Used. None.

Memory Required. 27 bytes.

Remarks. The approximation is most accurate when one of the tabulated x-values is greater than the desired x-value and the other is less.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] L.
3. Key in the variables at each prompt and press [R/S] .
4. See the y-value approximation.
5. For a new case, go to step 2.

Variables Used.

X = the x-value not found in the table.

A = the x-value of the first pair of tabulated values.

B = the x-value of the second pair.

C = the y-value of the first pair of tabulated values.

D = the y-value of the second pair.

Y = the corresponding y-value approximation.

Example. The saturation pressure of steam at 110°F is 1.2763 psi, and at 120 °F it is 1.6945 psi. What is the saturation pressure when the temperature is 113°F?

Keys:	Display:	Description:
[XEQ] L	X?*value*	
113 [R/S] 110 [R/S] 120 [R/S] 1.2763 [R/S]	A?*value* B?*value* C?*value* D?*value*	Inputs known values.
1.6945 [R/S]	Y=1.4018	Displays approximation of the saturation pressure.

Circle Determined by Three Points

This program calculates the center (x_0, y_0) and radius (r) of the circle defined by three noncollinear points.

$$r^2 = (x - x_0)^2 + (y - y_0)^2$$

Program Listing.

```
C01 LBL C
C02 INPUT A
C03 INPUT B
C04 INPUT C
C05 INPUT D
C06 INPUT E
C07 INPUT F
C08 RCL A
C09 STO- C
C10 STO- E
C11 RCL B
C12 STO- D
C13 STO- F
C14 RCL D
C15 RCL C
C16 y,x→θ,r
C17 STO X
C18 x<>y
C19 STO T
C20 RCL F
C21 RCL E
C22 y,x→θ,r
C23 x<>y
C24 RCL- T
C25 x<>y
C26 θ,r→y,x
C27 STO Z
C28 x<>y
C29 STO Y
C30 ÷
C31 RCL Z
C32 RCL- X
C33 ×
C34 RCL+ Y
C35 RCL X
C36 y,x→θ,r
C37 2
C38 ÷
C39 STO R
C40 RCL T
C41 0
C42 CMPLX+
C43 θ,r→y,x
C44 RCL B
C45 RCL A
C46 CMPLX+
C47 STO X
C48 x<>y
C49 STO Y
C50 VIEW X
C51 VIEW Y
C52 VIEW R
C53 RTN
Checksum = A34B
```

Flags Used. None.

Memory Required. 79.5 bytes.

Remarks. A divide-by-zero error occurs if the three points are collinear. The program modifies the variables that store x_2, y_2, x_3, and y_3; so if you repeat the program, you must reenter these values.

Program Instructions.

1. Key in the program listing; press [C] when finished.
2. Press [XEQ] C.
3. Key in the x- or y-coordinate (A through F) at each prompt and press [R/S].
4. After the y-coordinate of the third point is entered (with [R/S]), the x-coordinate of the center of the circle is displayed.
5. Press [R/S] and see the y-coordinate of the center.
6. Press [R/S] and see the radius of the circle.
7. For a new case, go to step 2.

Variables Used.

A = the x-coordinate of the first point.

B = the y-coordinate of the first point.

C = the x-coordinate of the second point.

D = the y-coordinate of the second point.

E = the x-coordinate of the third point.

F = the y-coordinate of the third point.

X = the x-coordinate of the center of the circle.

Y = the y-coordinate of the center of the circle.

R = the radius of the circle.

T, Z = variables used for intermediate results.

Example. Find the center and radius of the circle defined by the points (1,0), (2,4.5), and (−4.4,3).

Keys:	Display:	Description:
[XEQ] C	A?*value*	
1 [R/S]	B?*value*	Inputs coordinates.
0 [R/S]	C?*value*	
2 [R/S]	D?*value*	
4.5 [R/S]	E?*value*	
4.4 [+/−] [R/S]	F?*value*	
3 [R/S]	X= −0.9775	Displays the *x*-coordinate of the center.
[R/S]	Y=2.8005	Displays the *y*-coordinate of the center.
[R/S]	R=3.4283	Displays the radius.

Step-by-Step Solutions for Your HP-32S Calculator

Engineering Applications contains a variety of programs, examples, and solutions to show how you can easily solve your engineering problems.

- **Electrical Engineering**
  Reactance Chart • Impedance of a Ladder Network • Smith Chart Conversions • Transistor Amplifier Performance
- **Mechanical Engineering**
  Black Body Thermal Radiation • Ideal Gas Equation • Conduit Flow • Static Equivalent at a Point • Composit Section Properties • Soderberg's Equation for Fatigue
- **Civil Engineering**
  Mohr's Circle for Stress • Field Angle Traverse
- **Statistics**
  t Statistics • Chi-Square Evaluation • F Distribution • Analysis of Variance (One Way) • Binomial Distribution • Poisson Distribution
- **Mathematics**
  Triangle Solutions • Derivative of a Function • Linear Interpolation • Circle Determined by Three Points

HEWLETT PACKARD

Reorder Number
00032-90057

00032-90058 English
Printed in Canada 6/88

0 88698 00035 9